

Carlo Sintini e Costantino Mustacchio

Grafici di funzioni

con il C64, il Plus4 e il C16



Il piacere del computer



Il piacere del computer

Il piacere del computer

- 1 *Tom Rugg e Phil Feldman* 32 programmi con il PET
- 2 *Rich Didday* Intervista sul personal computer, hardware
- 3 *Tom Rugg e Phil Feldman* 32 programmi con l'Apple
- 4 *Ken Knecht* Microsoft Basic
- 5 *Paul M. Chirlian* Pascal
- 6 *Tom Rugg e Phil Feldman* 32 programmi con il TRS-80
- 7 *Rich Didday* Intervista sul personal computer, software
- 8 *Herbert D. Peckham* Imparate il Basic con il PET/CBM
- 9 *Karl Townsend e Merl Miller* Il personal computer come professione
- 10 *Karen Billings e David Moursund* Te ne intendi di computer?
- 11 *T. Dwyer e M. Critchfield* Il Basic e il personal computer, uno: introduzione
- 12 *Don Inman e Kurt Inman* Imparate il linguaggio dell'Apple
- 13 *T. Dwyer e M. Critchfield* Il Basic e il personal computer, due: applicazioni
- 14 *Luigi Pierro* Il manuale del CP/M
- 15 *Carlo Sintini* A scuola con il PET/CBM
- 16 *David Johnson-Davies* Il manuale dell'Atom
- 17 *David E. Schultz* Il libro del Commodore VIC 20
- 18 *Jim Huffman e Robert Bruce* Il "debug" nei personal computer
- 19 *John M. Nevison* Programmazione in Basic per l'uomo d'affari
- 20 *Mark Harrison* Imparate il Basic con lo ZX81
- 21 *Ronald W. Anderson* Dal Basic al Pascal
- 22 *Herbert D. Peckham* Imparate il Basic con il Texas TI 99/44
- 23 *Sergio Borsani* A scuola con il Texas TI 99/4A
- 24 *Jerry Willis e Deborah Willis* Come usare il Commodore 64
- 25 *Mark Harrison* Imparate il Basic con lo Spectrum
- 26 *Carlo Sintini e Costantino Mustacchio* A scuola con il Commodore 64
- 27 *David A. Lien* Imparate il Basic con l'IBM PC
- 28 *Ken Tracton* Introduzione al Lisp
- 29 *Fabio Mavaracchio* Programmi in Basic per l'elettronica
- 30 *Ian Stewart e Robin Jones* Il linguaggio macchina dello Spectrum
- 31 *Tom Rugg, Phil Feldman e C.S. Wilson* 32 programmi per il VIC 20
- 32 *Merl Miller e Mary A. Myers* Introduzione all'Apple Macintosh
- 33 *Stam Krute* Grafica e suoni con il Commodore 64
- 34 *Jerry Willis e William Manning* Come usare l'IBM PCjr
- 35 *Tom Rugg e Phil Feldmann* 32 programmi con il Commodore 64
- 36 *Sergio Borsani* Matematica e geometria con il Commodore 64
- 37 *David Laine* ZX Spectrum: tecniche avanzate di linguaggio macchina
- 38 *Salvatore Marseglia* Chimica con il pocket computer
- 39 *Patrizio Quintilli* Basic per i geometri
- 40 *Roy Atherton* Programmare in SuperBasic con il QL
- 41 *Carl Townsend* Il sistema operativo MSDOS

Carlo Sintini e Costantino Mustacchio

Grafici di funzioni

con il C64, il Plus4 e il C16



franco muzzio & c. editore

Direzione editoriale di Virginio Sala
Redazione tecnica di Sergio Fardin
Composizione e impaginazione del Centro Fotocomposizione Dorigo - Padova
Copertina di Elda Danese

Prima edizione settembre 1985
ISBN 88-7021-332-3

© 1985 franco muzzio & c. editore spa
Via Makallè 73, 35138 Padova, tel. 049/661147-661873
Tutti i diritti sono riservati

Indice generale

- 7 **Visualizzazione di una funzione in HI-RE con il C64**
Inserimento automatico della funzione nel listato Altra risoluzione e tracciamento Elaborazione e tracciamento punti Listato definitivo
- 20 **Stampa di una funzione con il C64**
Routine di hard copy Avvertenze per la stampa Elenco di funzioni da visualizzare. Esempi di utilizzazione
- 39 **Stampa di funzioni polidrome o sovrapposte con il C64**
Descrizione del listato Esempi di utilizzazione Applicazioni nella scomposizione di Fourier Funzioni reciproche Sovrapposizione fra due onde: Battimenti Figure di Lissajou
- 64 **Stampa di funzioni con il Simon's Basic**
Descrizione del listato Ricerca dei massimi e minimi di una funzione Grafico di una funzione e calcolo di un'area Esempi
- 80 **Grafici con il Plus 4 e C16**
Grafico di una funzione e calcolo di un'area Funzioni polivoche o sovrapposte con il C16 e il Plus 4 Funzioni in coordinate polari Numeri complessi e loro rappresentazione nel piano di Gauss

Visualizzazione di una funzione in HI-RES con il C64

INSERIMENTO AUTOMATICO DELLA FUNZIONE NEL LISTATO

La funzione di cui si deve generare il grafico non può essere introdotta per mezzo del comando `INPUT` perché questo accetta solo variabili numeriche o di stringa e non espressioni contenenti operatori aritmetici. In altre parole, se tentassimo di battere

```
100 INPUT "Y =" ;Y
```

potremmo introdurre in `Y` (o in `Y$`) un singolo valore numerico ma mai una espressione contenente una variabile e segni di operazione.

È necessario perciò usare il comando `DEF FNY(X)= ...`, ma sarebbe scomodo modificare manualmente il listato per introdurre di volta in volta la funzione che si vuole visualizzare sullo schermo.

La soluzione migliore consiste in una routine che ci permetta di introdurre automaticamente la funzione desiderata nel listato sotto la forma `DEF FNY(X)= ...`

Si provi a lanciare la seguente routine:

```
100 PRINT"(CLEAR)BATTI LA FUNZIONE :"  
110 INPUT"(3DOWN)Y = " ;Y$  
120 PRINT"(CLEAR)(3DOWN)500 DEF FNY(X)="Y$  
130 PRINT"GOTO 500(HOME)"
```

```
150 END
510 PRINT"LA FUNZIONE SI TROVA ORA ALLA LINEA 500"
```

Supponiamo per esempio di introdurre alla riga 110 la funzione:

$$x*x-7*(x-13)+4*x-3$$

sotto forma di stringa e memorizzata nella variabile Y\$.

Nelle righe 120 e 130 viene cancellato lo schermo e stampato il messaggio seguente:

```
2 RIGHE VUOTE }
POSIZIONE →   }
CURSORE       }
READY.
500 DEF FNY(X)=X*X-7*(X-13)+4*X-3
GOTO 500
```

Fig. 1.1

Infatti il computer stampa i messaggi contenuti in 120 e 130 (rispettivamente sulla quarta e quinta riga di schermo), poi il cursore viene riportato in HOME e viene stampato il messaggio READY di fine programma.

Se a questo punto battete per due volte consecutive il tasto RETURN vengono rese operative le scritte che passano sotto il cursore.

Quindi viene prima introdotta la riga 500 nel listato (o viene modificata nel caso in cui fosse già presente) e successivamente si riprende a scorrere dalla riga 500.

È facile verificare, con un LIST, che effettivamente la riga 500 è stata introdotta nel programma in modo definitivo.

Questo procedimento può essere migliorato forzando tre RETURN automatici ed eliminando così la necessità di eseguire manualmente tale operazione.

Il computer ha la capacità di memorizzare l'ordine con cui vengono

premuti più tasti in successione. La porzione di memoria adibita a tale impiego si chiama *buffer di tastiera* e inizia alla locazione 631.

Nella locazione 198 viene invece precisato quanti caratteri il computer deve prelevare dal buffer di tastiera e rendere operanti (fino ad un massimo di 10 caratteri).

A questo punto possiamo sfruttare il buffer di tastiera aggiungendo al listato precedente la riga

```
150 FOR K=0 TO 2:POKE 631+K,13:NEXT:POKE 198,3
```

Grazie al ciclo FOR...NEXT vengono introdotti tre RETURN (CHR\$(13)=RETURN) nel buffer di tastiera e quindi viene precisato al calcolatore di prelevare tre caratteri dal buffer (infatti nella locazione 198 viene detto al calcolatore quanti caratteri deve prelevare dal buffer di tastiera). Abbiamo aggiunto un RETURN più del necessario per maggior sicurezza e per rendere operativa una eventuale altra riga.

Provate a far girare la routine e vi accorgete di quanto essa sia efficace e suscettibile di applicazioni anche per altri impieghi.

È buona norma riportare la locazione 198 in condizioni normali dopo aver sfruttato queste routine, con il comando

```
POKE 198,0
```

Finalmente siamo in grado di realizzare una routine definitiva che ci permette di introdurre nel listato la funzione da presentare in grafico, di stabilire gli intervalli di visualizzazione e di precisare la presenza di eventuali asintoti verticali.

La necessità di segnalare al computer la presenza degli asintoti nasce dal fatto che nelle ascisse corrispondenti l'ordinata è infinita e perciò nel calcolo dell'ordinata stessa il calcolatore incorrerebbe nell'errore di divisione per zero o nell'errore di *overflow*.

Conoscendo invece tali ascisse eviteremo il calcolo dell'ordinata facendo saltare il programma al tracciamento dell'asintoto.

```
100 CLR:PRINT"[CLEAR][BLEU]":POKE 53281,15:POKE 53280,8
110 PRINT"[CLEAR]BATTI LA FUNZIONE : "
120 INPUT "[3 DOWN]Y = ";Y$
130 PRINT"[CLEAR][3 DOWN]500 DEF FNY(X)="Y$
140 PRINT"GOTO 500[HOME]"
150 FOR K=0 TO 2:POKE 631+K,13:NEXT:POKE 198,3
160 END
505 POKE 198,0:PRINT"[CLEAR]OK - OKA ";
510 PRINT"SCEGLI GLI INTERVALLI DI "
520 PRINT TAB(5)"VISUALIZZAZIONE : "
530 INPUT "[2 DOWN]X (SINISTRA)=";X1
540 INPUT "X (DESTRA)=";X2
550 IF X1>X2 THEN 510
560 INPUT "[DOWN]Y (BASSA)=";Y1
```

```

570 INPUT "Y (ALTA)=";Y2
580 IF Y1>=Y2 THEN 510
590 PRINT"[DOWN]CI SONO ASINTOTI VERTICALI ? ";
600 GOSUB 15000:PRINT Q$:IF Q$="N" THEN 660
610 PRINT"[DOWN]QUANTI ? ";
620 GOSUB 15000:PRINT Q$:NA=Q
630 PRINT:FOR K=1 TO NA
640 PRINT"ASCISSA"K"" ASINTOTO =";:INPUT AS(K)
650 NEXT K
660 END
15000 REM -----
15010 REM          ROUTINE DI ATTESA
15020 REM -----
15030 GET Q$:IF Q$="" THEN 15030
15040 Q=VAL(Q$):RETURN

```

Non abbiamo ritenuto necessario dimensionare la variabile AS(K), perché consideriamo assolutamente improbabile che sia necessario visualizzare più di 10 asintoti verticali contemporaneamente e, come è noto, nei computer Commodore non è necessario dimensionare le variabili con indice quando il dimensionamento non supera il valore dieci.

La subroutine 15000 serve per creare un ciclo che ... tenga impegnato il computer in attesa che venga premuto un tasto.

ALTA RISOLUZIONE E TRACCIAMENTO

Ogni carattere che appare sullo schermo è costituito da una matrice di 8 × 8 punti (o pixel) opportunamente accesi o spenti (cioè con le stesse dimensioni del cursore lampeggiante).

In condizioni normali lo schermo contiene mille caratteri e quindi è strutturato con:

$$8 \times 8 \times 1000 = 64000 \text{ punti}$$

Si può accedere all'alta risoluzione, cioè alla gestione dei singoli pixel, per mezzo della seguente routine:

```

100 POKE 53272,PEEK(53272) OR 8
110 POKE 53265,PEEK(53265) OR 32
120 FOR K=8192 TO 8192+7999
130 POKE K,0
140 NEXT K
150 FOR K=1024 TO 2023
160 POKE K,3

```

170 NEXT K
180 RETURN

Nella riga 100 viene forzato un 1 nel quarto bit della locazione 53272 (registro interno del vic che controlla l'inizio della matrice video).

In 110 viene forzato un 1 nel sesto bit della locazione 53265 (registro interno del vic adibito a interruttore per attivare l'alta risoluzione).

Nel primo ciclo FOR...NEXT viene pulito tutto l'intervallo di memoria riservato alla memoria di schermo in alta risoluzione. Essa inizia alla locazione 8192 e termina dopo 8000 byte, in quanto ogni locazione raccoglie i dati di otto pixel (1 pixel per ciascun bit).

Infine con il secondo ciclo FOR...NEXT coloriamo in cyan lo sfondo dello schermo. Infatti in alta risoluzione l'intervallo di memoria che comunemente è riservato ai caratteri di schermo viene trasformato in memoria del colore.

Si noti come, per questa ragione, non è possibile inserire più di un colore all'interno di un quadratino di 8×8 pixel.

Questa routine ha lo svantaggio di essere piuttosto lenta ed è quindi opportuno sostituirla con una identica subroutine in linguaggio macchina.

```

20000 REM -----
20010 REM  ROUTINE PER ATTIVAZ. HI-RES
20020 REM -----
20030 FOR K= 49152 TO 49222
20040 READ N
20050 POKE K,N
20070 NEXT K
20080 RETURN
20090 DATA 173,24,208,9,8,141,24,208,173,17,208,9,32,141,17,208
      .169,0,133
20100 DATA 251,169,32,133,252,160,0,169,0,145,251,200,192,0,208
      .249,230
20110 DATA 252,169,64,197,252,208,239,169,0,133,251,169,4,133,2
      52,160,0
20120 DATA 169,3,145,251,200,192,0,208,249,230,252,169,8,197,25
      2,208,239,96

```

Con l'attivazione dell'alta risoluzione possiamo ora operare su un reticolato di pixel ordinati nel modo seguente.




Fig. 1.2

Il byte zero è quello corrispondente alla locazione 8192. Si noti come il primo pixel che si trova nell'angolo superiore sinistro dello schermo corrisponde al bit più alto del byte zero mentre l'ultimo pixel in basso a destra corrisponde al bit più basso della locazione $8192 + 7999$.

Le coordinate di ciascun pixel sono fissate con riferimento all'angolo superiore sinistro dello schermo. Perciò il pixel di coordinate 100,50 corrisponde al punto che si ottiene partendo dall'angolo suddetto e spostandoci verso destra di cento pixel e verso il basso di 50.

Osservando la figura risulta evidente la disposizione apparentemente innaturale con cui procedono i pixel mano a mano che le ascisse o le ordinate aumentano. È necessario perciò strutturare comandi che facilitino l'individuazione del pixel da accendere.

Supponiamo di voler accendere un unico pixel di coordinate x, y .

INT (Y/8)	Serve per determinare a quale delle 25 righe appartiene il pixel.
INT (X/8)	Serve per determinare a quale delle 40 colonne appartiene il pixel.
Y AND 7	Serve per determinare a quale degli otto byte (della matrice quadrata di 8×8 bit) appartiene il pixel.

7-(X AND 7) Serve per determinare a quale bit del byte appartiene il pixel.

Per l'accensione del pixel desiderato possiamo quindi usare la subroutine seguente:

```
25030 IF X<0 OR X>319 OR Y<0 OR Y>199 THEN 25060
25040 BY=8192+320*INT(Y/8)+8*INT(X/8)+(Y AND 7)
25050 POKE BY,PEEK(BY) OR 2*(7-(X AND 7))
25060 RETURN
```

In 25030 viene effettuato un controllo per evitare l'uso di coordinate illegali.

Nella riga 25040 si determina in quale byte è contenuto il bit corrispondente al pixel da accendere.

Infine nella riga 25050 viene forzato un 1 in corrispondenza del bit desiderato.

Per disattivare l'alta risoluzione e tornare in bassa risoluzione, usate i due comandi seguenti:

```
30030 POKE 53272,PEEK(53272) AND 247
30040 POKE 53265,PEEK(53265) AND 223
```

dove 247 e 223 sono rispettivamente i complementi di 8 e 32 che con l'operazione logica AND forzano uno zero nei due registri già visti nella fase di attivazione dell'alta risoluzione e che erano stati portati a 1. Abbiamo così ristabilito le condizioni iniziali e siamo tornati in bassa risoluzione.

Il listato precedente può quindi essere integrato nel modo seguente

```
100 GOSUB 20000:X=160:Y=100:SYS49152:GOSUB 25000
110 GET Q$:IF Q$="" THEN 110
120 GOSUB 30000:END
20000 REM -----
20010 REM  ROUTINE PER ATTIVAZ. HI-RES
20020 REM -----
20030 FOR K= 49152 TO 49222
20040 READ N
20050 POKE K,N
20070 NEXT K
20080 RETURN
20090 DATA 173,24,208,9,8,141,24,208,173,17,208,9,32,141,17,208
.169,0,133
20100 DATA 251,169,32,133,252,160,0,169,0,145,251,200,192,0,208
.249,230
20110 DATA 252,169,64,197,252,208,239,169,0,133,251,169,4,133,2
52,160,0
20120 DATA 169,3,145,251,200,192,0,208,249,230,252,169,8,197,25
2,208,239,96
```

```

25000 REM -----
25010 REM      ROUTINE PER COMANDO PLOT
25020 REM -----
25030 IF X<0 OR X>319 OR Y<0 OR Y>199 THEN 25060
25040 BY=8192+320*INT(Y/8)+8*INT(X/8)+(Y AND 7)
25050 POKE BY,PEEK(BY) OR 2*(7-(X AND 7))
25060 RETURN
30000 REM -----
30010 REM      ROUTINE PER DISATTIV. HI-RES
30020 REM -----
30030 POKE 53272,PEEK(53272) AND 247
30040 POKE 53265,PEEK(53265) AND 223
30050 PRINT"[CLEAR]":RETURN

```

Le righe 100, 110 e 120 servono solo per controllare il buon funzionamento delle subroutine. Se tutto è stato digitato correttamente, si dovrà accendere un unico pixel al centro dello schermo (coordinate 160, 100).

ELABORAZIONE E TRACCIAMENTO PUNTI

Con l'alta risoluzione lo schermo viene trasformato in una griglia di 64000 punti disposti su 200 righe e 320 colonne. Ciascuno di questi punti viene chiamato pixel e il pixel di coordinate 0,0 corrisponde all'angolo superiore sinistro dello schermo. L'ultimo pixel è invece quello con coordinate 319, 199 e corrisponde all'angolo inferiore destro.

Ci occuperemo di funzioni (algebriche o trascendenti) del tipo:

$$y = f(x)$$

cioè funzioni univoche con al massimo un solo valore per ogni ascissa. Il grafico della funzione si otterrà con un ciclo FOR...NEXT per le ascisse che vanno da zero a 319. Per ciascuna di queste ascisse sarà calcolata la corrispondente ordinata e poi questa verrà visualizzata con l'accensione del relativo pixel.

Un primo problema importante da chiarire è la differenza tra coordinate *assolute* e *relative*. Infatti noi scegliamo un opportuno intervallo di visualizzazione sia per le ascisse che per le ordinate, aprendo così un'ipotetica "finestra" entro la quale avremo la visualizzazione della funzione. Le coordinate relative a tale finestra saranno appunto le *coordinate relative* e generalmente queste saranno differenti dalle coordinate effettive o *assolute* della funzione.

Indicheremo con x_1, x_2, y_1, y_2 gli estremi della finestra, come indicato in figura.




Fig. 1.3

Per esempio il punto *A* della parabola ha coordinate assolute 1,2 e coordinate relative 160,100.

Il passaggio da coordinate assolute (x_0, y_0) a coordinate relative (x, y) si ottiene con le formule seguenti:

$$x = (x_0 - x_1)/Dx$$

$$y = (y_0 - y_1)/Dy$$

dove x e y rappresentano rispettivamente il numero di unità assolute che intercorrono fra un pixel e il successivo

$$Dx = (x_2 - x_1)/320$$

$$Dy = (y_2 - y_1)/200$$

Le formule inverse per passare da coordinate relative a coordinate assolute sono invece:

$$x_0 = x_1 + xDx$$

$$y_0 = y_1 + yDy$$

Una volta fissati gli intervalli di visualizzazione, potremo quindi calcolare le coordinate relative da riportare in grafico con la routine seguente.

```

1030 .SYS 49152 : REM ATTIVAZIONE HI-RES
1040 DX = (X2-X1)/320 : DY = (Y1-Y2)/200
1060 FOR X=0 TO 319
1070 X0=X1+X*DX
1080 Y0=FNY(X0)
1090 IF Y0<=Y1 THEN 1130
1100 IF Y0>=Y2 THEN 1130
1110 Y=200-(Y0-Y1)/DY
1120 GOSUB 25000: REM SUBROUTINE DI PLOTTAMENTO
1130 NEXT X
1140 GOSUB 15000: REM SUBROUTINE DI ATTESA
1150 GOSUB 30000: REM SUBROUTINE DI RITORNO IN BASSA
      RISOLUZIONE

```

Nella riga 1110 viene prima calcolata l'ordinata relativa $((Y0-Y1)/DY)$ e quindi viene determinato il complemento a 200 perché nelle coordinate assolute la Y aumenta andando verso l'alto, mentre in quelle relative la Y aumenta andando verso il basso.

Le righe 1090 e 1100 servono a saltare la routine di tracciamento nei casi in cui l'ordinata non risulti dentro la finestra di visualizzazione.

LISTATO DEFINITIVO

Possiamo ora procedere alla elaborazione del listato finale.

Nella parte introduttiva (fino alla riga 650) viene introdotta la funzione, vengono scelti gli intervalli di visualizzazione e viene notificata al computer l'ascissa relativa agli asintoti verticali.

Si passa alla routine di elaborazione con una SYS di attivazione dell'alta risoluzione, e con la determinazione dei parametri X e Y.

Nella riga 1050 si ha l'invio al tracciamento degli assi cartesiani.

Nelle righe comprese fra la 1060 e la 1130, con un ciclo FOR...NEXT si prendono in considerazione tutte le ascisse relative comprese tra 0 e 319; per ciascuna viene determinato il corrispondente valore assoluto X0 (riga 1070) e si ha il passaggio per la subroutine 13000 in cui si ha un test per controllare se tale ascissa assoluta coincide con uno degli asintoti. In caso affermativo si passa alla riga 13060 per il tracciamento dell'asintoto stesso, l'incremento dell'ascissa relativa, il calcolo della nuova ascissa assoluta e il ritorno dalla subroutine (riga 13100).

In caso negativo si ha il ritorno immediato nella linea 13050.

Ritornando nel programma principale in 1080, si ha il calcolo dell'ordinata assoluta mentre, nelle due linee successive viene controllato se tale

ordinata cade nella finestra della visualizzazione. In caso negativo si ha il salto alla riga 1130. In caso affermativo nella riga 1110 viene calcolata l'ordinata relativa, più precisamente il suo complemento a 200. Si passa quindi alla subroutine 25000 di tracciamento.

Infine in 15000 si ha la routine di attesa e in 30000 la disattivazione dell'alta risoluzione.

```

100 CLR :PRINT"[CLEAR][BLEU]":POKE 53281,15:POKE 53280,8:GOS
  UB 20000
110 PRINT"[CLEAR]BATTI LA FUNZIONE :"
120 INPUT "[3 DOWN]Y = ";Y$
130 PRINT"[CLEAR][3 DOWN]500 DEF FNY(X)="Y$
140 PRINT"GOTO 500[HOME]"
150 FOR K=0 TO 2:POKE 631+K,13:NEXT:POKE 198,3
160 END
500 DEF FNY(X)=(X-2)/(X+2)
505 POKE 198,0:PRINT"[CLEAR]OK - ORA "
510 PRINT"SCEGLI GLI INTERVALLI DI"
520 PRINT TAB(5)"VISUALIZZAZIONE :"
530 INPUT "[2 DOWN]X (SINISTRA)=":X1
540 INPUT "X (DESTRA)=":X2
550 IF X1>X2 THEN 510
560 INPUT "[DOWN]Y (BASSA)=":Y1
570 INPUT "Y (ALTA)=":Y2
580 IF Y1>Y2 THEN 510
590 PRINT"[DOWN]CI SONO ASINTOTI VERTICALI ? "
600 GOSUB 15000:PRINT Q$:IF Q$="N" THEN 1000
610 PRINT"[DOWN]QUANTI ? "
620 GOSUB 15000:PRINT Q$:NA=Q
630 PRINT:FOR K=1 TO NA
640 PRINT"ASCISSA"K"" ASINTOTO =":INPUT AS(K)
650 NEXT K
1000 REM -----
1010 REM ELABORAZIONE
1020 REM -----
1030 SYS 49152
1040 DX=(X2-X1)/320:DY=(Y2-Y1)/200
1050 GOSUB 10000:GOSUB 11000
1060 FOR X=0 TO 319
1070 X0=X1+X*DX:GOSUB 13000
1080 Y0=FNY(X0)
1090 IF Y0<=Y1 THEN 1130
1100 IF Y0>=Y2 THEN 1130
1110 Y=200-(Y0-Y1)/DY
1120 GOSUB 25000
1130 NEXT X
1140 GOSUB 15000
1150 GOSUB 30000
2000 REM -----
2010 REM MESSAGGI FINALI
2020 REM -----
2030 PRINT"VUOI CAMBIARE GLI INTERVALLI DI"
2040 PRINT"VISUALIZZAZIONE ?"
2050 GOSUB 15000
2060 IF Q$="S" THEN PRINT"[CLEAR]":GOTO 510
2070 PRINT"[DOWN]VUOI CAMBIARE LA FUNZIONE ?"
2080 GOSUB 15000
2090 IF Q$="S" THEN RUN
2100 PRINT"[DOWN]OK - FINE LAVORO."
2110 END
10000 REM -----

```

```

10010 REM          ASSE X
10020 REM -----
10030 IF SGN(Y1)=SGN(Y2) THEN RETURN
10040 Y=INT(Y2/DY+.5)
10050 FOR X=0 TO 319 STEP 6
10060 GOSUB 25000
10070 NEXT X:RETURN
11000 REM -----
11010 REM          ASSE Y
11020 REM -----
11030 IF SGN(X1)=SGN(X2) THEN RETURN
11040 X=-INT(X1/DX+.5)
11050 FOR Y=0 TO 199 STEP 6
11060 GOSUB 25000
11070 NEXT Y:RETURN
13000 REM -----
13010 REM          ASINTOTI
13020 REM -----
13030 FOR J=1 TO NA
13040 IF X0=AS(J) THEN 13060
13050 NEXT J:RETURN
13060 FOR Y=0 TO 199 STEP 3
13070 GOSUB 25000
13080 NEXT Y
13090 X=X+1:X0=X1+X*DX
13100 RETURN
15000 REM -----
15010 REM          ROUTINE DI ATTESA
15020 REM -----
15030 GET Q$:IF Q$="" THEN 15030
15040 Q=VAL(Q$):RETURN
20000 REM -----
20010 REM          ROUTINE PER ATTIVAZ. HI-RES
20020 REM -----
20030 FOR K= 49152 TO 49222
20040 READ N
20050 POKE K,N
20070 NEXT K
20080 RETURN
20090 DATA 173,24,208,9,8,141,24,208,173,17,208,9,32,141,17,208
,169,0,133
20100 DATA 251,169,32,133,252,160,0,169,0,145,251,200,192,0,208
,249,230
20110 DATA 252,169,64,197,252,208,239,169,0,133,251,169,4,133,2
52,160,0
20120 DATA 169,3,145,251,200,192,0,208,249,230,252,169,8,197,25
2,208,239,96
25000 REM -----
25010 REM          ROUTINE PER COMANDO PLOT
25020 REM -----
25030 IF X<0 OR X>319 OR Y<0 OR Y>199 THEN 25060
25040 BY=8192+320*INT(Y/8)+8*INT(X/8)+(Y AND 7)
25050 POKE BY,PEEK(BY) OR 21<(7-(X AND 7))
25060 RETURN
30000 REM -----
30010 REM          ROUTINE PER DISATTIV. HI-RES
30020 REM -----
30030 POKE 53272,PEEK(53272) AND 247
30040 POKE 53265,PEEK(53265) AND 223
30050 PRINT"[CLEAR]":RETURN

```

Per il tracciamento degli assi coordinati nelle apposite subroutine 10000 e 11000 si ha prima il controllo che i segni dei due estremi di visualizza-

zione sono differenti: infatti se essi sono concordi l'asse opposto cade sicuramente fuori della finestra e non deve essere visualizzato.

In tal caso si ha il RETURN immediato mentre, in caso contrario, viene calcolata la seconda coordinata relativa e con un ciclo FOR...NEXT e l'uso della subroutine 25000 viene realizzato il tracciamento dell'asse in questione.

Il programma principale termina con le righe da 2000 a 2110 contenenti le richieste finali.

In caso si vogliono cambiare gli intervalli di visualizzazione viene evitato il passaggio di introduzione della funzione e di lettura dei DATA occorrenti per la routine dell'alta risoluzione.

Stampa di una funzione con il C64

Le funzioni visualizzate sullo schermo possono essere trasferite su carta con una opportuna routine in linguaggio macchina, sistemata in memoria a partire dalla locazione 52224 fino alla locazione 52667, e che funziona con la stampante MPS-801 della Commodore.

La stampa viene eseguita trasversalmente nello stesso verso di scorrimento del foglio, ed è possibile ottenere una dilatazione o meno lungo l'asse x con i comandi:

POKE 2,2 stampa amplificata

POKE 2,1 stampa ristretta

```
100 CLR :PRINT"[CLEAR][BLEU]":POKE 53281,15:POKE 53280,8
105 PRINTTAB(7)"[2 DOWN]ATTENDERE PER FAVORE .....":GOSUB 200
    00:GOSUB 32100
110 PRINT"[CLEAR]BATTI LA FUNZIONE : "
120 INPUT "[3 DOWN]Y = ";Y#
130 PRINT"[CLEAR][3 DOWN]500 DEF FNY(X)="Y#
140 PRINT"GOTO 500[HOME]"
150 FOR K=0 TO 2:POKE 631+K,13:NEXT:POKE 198,3
160 END
500 DEF FNY(X)=2*SIN(X)+(COS(X)/2)
505 POKE 198,0:PRINT"[CLEAR]OK - ORA "
510 PRINT"SCEGLI GLI INTERVALLI DI "
520 PRINT TAB(5)"VISUALIZZAZIONE : "
530 INPUT "[2 DOWN]X (SINISTRA)=";X1
540 INPUT "X (DESTRA)=";X2
550 IF X1>X2 THEN 510
560 INPUT "[DOWN]Y (BASSA)=";Y1
```

```

570 INPUT "Y (ALTRA)=";Y2
580 IF Y1=Y2 THEN 510
590 PRINT"[DOWN]CI SONO ASINTOTI VERTICALI ? ";
600 GOSUB 15000 PRINTQ$;IF Q$="N" THEN NA=-1 GOTO 1000
610 PRINT"[DOWN]QUANTI ? ";
620 GOSUB 15000 PRINTQ$;NA=Q
630 PRINT:FOR K=1 TO NA
640 PRINT"ASCISSA"K"* ASINTOTO =";:INPUT AS(K)
650 NEXT K
1000 REM -----
1010 REM          ELABORAZIONE
1020 REM -----
1030 SYS 49152
1040 DX=(X2-X1)/320;DY=(Y2-Y1)/200
1050 GOSUB 10000 GOSUB 11000
1060 FOR X=0 TO 319
1070 X0=X1+X*DX;GOSUB 13000
1080 Y0=FN(X0)
1090 IF Y0<Y1 THEN 1130
1100 IF Y0=Y2 THEN 1130
1110 Y=200-(Y0-Y1)/DY
1120 GOSUB 25000
1130 NEXT X
1140 GOSUB 15000
1145 IF Q$="+" THEN GOSUB 32000
1150 GOSUB 30000
2000 REM -----
2010 REM          MESSAGGI FINALI
2020 REM -----
2030 PRINT"VUOI CAMBIARE GLI INTERVALLI DI"
2040 PRINT"VISUALIZZAZIONE ?";
2050 GOSUB 15000
2060 IF Q$="S" THEN PRINT"[CLEAR]";:GOTO 510
2070 PRINT"[DOWN]VUOI CAMBIARE LA FUNZIONE ?";
2080 GOSUB 15000
2090 IF Q$="S" THEN RUN
2100 PRINT"[DOWN]OK - FINE LAVORO."
2110 END
10000 REM -----
10010 REM          ASSE X
10020 REM -----
10030 IF SGN(Y1)=SGN(Y2) THEN RETURN
10040 Y=INT(Y2/DY+.5)
10050 FOR X=0 TO 319 STEP 6
10060 GOSUB 25000
10070 NEXT X:RETURN
11000 REM -----
11010 REM          ASSE Y
11020 REM -----
11030 IF SGN(X1)=SGN(X2) THEN RETURN
11040 X=-INT(X1/DX+.5)
11050 FOR Y=0 TO 199 STEP 6
11060 GOSUB 25000
11070 NEXT Y:RETURN
13000 REM -----
13010 REM          ASINTOTI
13020 REM -----
13025 IF NA=-1 THEN 13100
13030 FOR J=1 TO NA
13040 IF X0=AS(J) THEN 13060
13050 NEXT J:RETURN
13060 FOR Y=0 TO 199 STEP 3
13070 GOSUB 25000
13080 NEXT Y
13090 X=X1+X0=X1+X*DX

```

```

13100 RETURN
15000 REM -----
15010 REM          ROUTINE DI ATTESA
15020 REM -----
15030 GET Q$:IF Q$="" THEN 15030
15040 Q=VAL(Q$):RETURN
20000 REM -----
20010 REM  ROUTINE PER ATTIVAZ. HI-RES
20020 REM -----
20030 FOR K= 49152 TO 49222
20040 READ N
20050 POKE K,N:ND=ND+N
20070 NEXT K
20080 RETURN
20090 DATA 173,24,208,9,8,141,24,208,173,17,208,9,32,141,17,208
,169,0,133
20100 DATA 251,169,32,133,252,160,0,169,0,145,251,200,192,0,208
,249,230
20110 DATA 252,169,64,197,252,208,239,169,0,133,251,169,4,133,2
52,160,0
20120 DATA 169,3,145,251,200,192,0,208,249,230,252,169,8,197,25
2,208,239,96
25000 REM -----
25010 REM          ROUTINE PER COMANDO PLOT
25020 REM -----
25030 IF X<0 OR X>319 OR Y<0 OR Y>199 THEN 25060
25040 BY=8192+320*INT(Y/8)+8*INT(X/8)+(Y AND 7)
25050 POKE BY,PEEK(BY) OR 2*(7-(X AND 7))
25060 RETURN
30000 REM -----
30010 REM  ROUTINE PER DISATTIV. HI-RES
30020 REM -----
30030 POKE 53272,PEEK(53272) AND 247
30040 POKE 53265,PEEK(53265) AND 223
30050 PRINT"[CLEAR]":RETURN
32000 REM -----
32010 REM          S T A M P A
32020 REM -----
32030 FOR X=0 TO 319 STEP 2:Y=0:GOSUB 25000:Y=199:GOSUB 2
5000:NEXT X
32040 FOR Y=0 TO 199 STEP 2:X=0:GOSUB 25000:X=319:GOSUB 2
5000:NEXT Y
32050 SYS 52224
32060 RETURN
32070 REM -----
32080 REM          ROUTINE PER STAMPA
32090 REM -----
32100 POKE 2,2:REM PER STAMPA RISTRETTA SOSTITUIRE CON POKE 2,
1
32110 FOR K=52224 TO 52677
32120 READ N:POKE K,N:ND=ND+N:NEXT K
32130 IF ND<>70923 THEN PRINT"(DOWN)ERRORE NEI DATA":END
32140 RETURN
32270 DATA 173,0,221,41,3,73
32280 DATA 3,160,6,10,136,208
32290 DATA 252,141,132,3,173,24
32300 DATA 208,41,8,240,9,24
32310 DATA 169,32,109,132,3,141
32320 DATA 132,3,169,0,32,189
32330 DATA 255,169,4,170,160,255
32340 DATA 32,186,255,32,192,255
32350 DATA 162,4,32,201,255,176
32360 DATA 3,76,61,204,76,32
32370 DATA 205,169,8,32,210,255
32380 DATA 169,13,32,210,255,162

```

32390 DATA 0,169,1,141,198,205
 32400 DATA 169,0,141,199,205,169
 32410 DATA 0,141,200,205,169,199
 32420 DATA 141,201,205,32,225,255
 32430 DATA 208,3,76,32,205,138
 32440 DATA 72,152,72,32,41,205
 32450 DATA 104,168,104,170,173,205
 32460 DATA 205,45,203,205,240,12
 32470 DATA 173,202,205,13,198,205
 32480 DATA 141,202,205,76,143,204
 32490 DATA 173,198,205,73,255,45
 32500 DATA 202,205,141,202,205,14
 32510 DATA 198,205,173,198,205,201
 32520 DATA 128,240,20,24,173,199
 32530 DATA 205,105,1,141,199,205
 32540 DATA 173,200,205,105,0,141
 32550 DATA 200,205,76,93,204,173
 32560 DATA 202,205,9,128,224,45
 32570 DATA 144,10,173,202,205,41
 32580 DATA 31,9,128,141,202,205
 32590 DATA 142,207,205,166,2,142
 32600 DATA 206,205,168,32,210,255
 32610 DATA 152,206,206,205,208,246
 32620 DATA 174,207,205,169,1,141
 32630 DATA 198,205,169,0,141,202
 32640 DATA 205,56,173,199,205,233
 32650 DATA 6,141,199,205,173,200
 32660 DATA 205,233,0,141,200,205
 32670 DATA 206,201,205,173,201,205
 32680 DATA 201,255,240,3,76,93
 32690 DATA 204,224,45,176,31,24
 32700 DATA 173,199,205,105,7,141
 32710 DATA 199,205,173,200,205,105
 32720 DATA 0,141,200,205,232,169
 32730 DATA 199,141,201,205,169,13
 32740 DATA 32,210,255,76,93,204
 32750 DATA 169,13,32,210,255,32
 32760 DATA 231,255,96,173,201,205
 32770 DATA 41,7,141,204,205,173
 32780 DATA 201,205,74,74,74,168
 32790 DATA 185,146,205,133,251,185
 32800 DATA 172,205,133,252,24,165
 32810 DATA 251,109,204,205,133,251
 32820 DATA 165,252,105,0,133,252
 32830 DATA 24,173,132,3,101,252
 32840 DATA 133,252,173,199,205,41
 32850 DATA 7,73,7,168,200,169
 32860 DATA 0,56,42,136,200,252
 32870 DATA 141,203,205,24,173,200
 32880 DATA 205,101,252,133,252,173
 32890 DATA 199,205,41,248,168,138
 32900 DATA 72,120,162,52,134,1
 32910 DATA 177,251,162,55,134,1
 32920 DATA 88,168,104,170,152,45
 32930 DATA 203,205,141,205,205,96
 32940 DATA 0,64,128,192,0,64
 32950 DATA 128,192,0,64,128,192
 32960 DATA 0,64,128,192,0,64
 32970 DATA 128,192,0,64,128,192
 32980 DATA 0,64,0,1,2,3
 32990 DATA 5,6,7,8,10,11
 33000 DATA 12,13,15,16,17,18
 33010 DATA 20,21,22,23,25,26
 33020 DATA 27,28,30,31

Dopo il tracciamento della funzione sullo schermo (riga 1140) si ha l'invio nella routine d'attesa e nel caso che venga premuto il tasto ← viene eseguito un salto alla subroutine di riga 32000 per l'esecuzione della stampa.

Nelle righe 32030 - 32040 viene tracciata una cornice, il cui scopo è delimitare sulla carta la finestra di visualizzazione della funzione.

In 32050 viene attivata la routine di *hard copy*, e infine si ha il RETURN per i messaggi finali.

Si noti che talvolta il programma sembra fermarsi inspiegabilmente per alcuni secondi quando l'ascissa assume il valore zero; ciò è dovuto a una confusione di variabili che si verifica quando non ci sono asintoti verticali.

Questo inconveniente può essere eliminato ponendo $NA = -1$ (riga 600) nel caso in cui non vi siano asintoti e aggiungendo la riga 13025 che realizza un salto impedendo il tracciamento dell'asintoto in corrispondenza dell'asse y .

Talvolta gli asintoti verticali possono non essere tracciati, e questo avviene quando a partire dall'ascissa minima x_1 con incrementi successivi Δx l'asintoto stesso viene a trovarsi in una posizione intermedia.

Un aspetto importante da sottolineare è l'impossibilità (per esempio $y = \text{tg}(x)$; $Y = \text{sec}(X)$; $Y = X/(X^2 - 3)$...) di segnalare per mezzo dell'INPUT asintoti verticali aventi per ascissa numeri irrazionali ($X = \pi/2$; $X = \text{SQR}(3)$...).

Per fortuna non è un problema: partendo dall'ascissa forzatamente razionale x_1 e con incrementi anche essi razionali, non si potrà mai pervenire all'elaborazione di una ascissa irrazionale. Quindi in ogni caso tali asintoti verticali verranno ignorati senza che sia necessario notificarli al calcolatore prima del tracciamento.

AVVERTENZE PER LA STAMPA

Il programma traccia la funzione con la massima precisione possibile senza tralasciare alcun punto e quindi con una velocità che talvolta può essere desiderabile aumentare.

La cornice eseguita all'inizio della fase di stampa ha lo scopo di delimitare il campo di tracciamento su carta con un effetto che riteniamo esteticamente valido.

È però possibile velocizzare il tracciamento del grafico eliminando completamente tale cornice, sopprimendo le righe 32030 - 32040, oppure sostituendole con:

```

32030 FOR X=0 TO 319 STEP 6:Y=0:GOSUB
      25000:Y=199:GOSUB 25000:NEXT X
32040 FOR Y=0 TO 199 STEP 6:X=0:GOSUB
      25000:X=319:GOSUB 25000:NEXT Y

```

Anche la funzione può essere descritta più rapidamente con un opportuno STEP, per esempio:

```
1060 FOR X=0 TO 319 STEP 3
```

Con lo stesso criterio si può aumentare la velocità di tracciamento degli assi (righe 10050 e 11050) e degli asintoti (riga 13060).

Prima di passare alla rassegna di alcuni esempi riteniamo utile segnalare una piccola anomalia caratteristica del sistema operativo.

Con il comando $B=INT(A)$ viene calcolata la parte intera nel numero memorizzato nella variabile A. Però, nel caso tale numero sia positivo, per esempio $A=3.5$, si avrà $B=3$; invece nel caso in cui tale numero sia negativo, per esempio $A=-3.5$, si avrà $B=-4$.

Infatti considerando i numeri reali in corrispondenza di una retta orientata verso destra il sistema operativo esegue in entrambi i casi l'approssimazione verso sinistra.




Fig. 2.1

ELENCO DI FUNZIONI DA VISUALIZZARE

Presentiamo, qui di seguito, una breve raccolta di funzioni di vario tipo che il lettore potrà sperimentare scegliendo di volta in volta gli opportuni intervalli.

Funzioni trigonometriche

(p sta per pi greco)

$$\begin{aligned}
y &= \sin(p/x) \\
y &= \sin(x)/x \\
y &= \tan(x)/(2*\sin(x)-1) \\
y &= x*\sin(x)/(1-\cos(x)) \\
y &= (1-\cos(x))/(3*\sin(x)*\sin(x)) \\
y &= 2*\cos(x)-\cos(2*x) \\
y &= (x \uparrow 2 - \sin(x))/(\cos(x)-1) \\
y &= (\sin x) \uparrow 3 + (\cos(x)) \uparrow 3 \\
y &= (\sin(x/2)) \uparrow 2 \\
y &= 2*\tan(x)/(1+(\tan(x)) \uparrow 2) \\
y &= (1+(\tan(x)) \uparrow 2)/(1-(\tan(x)) \uparrow 2) \\
y &= \sin(x)/(1+\cos(x)) \\
y &= \cos(2*x)+3*\cos(x) \\
y &= (\sin(x)-2)/(3*\sin(x)-1) \\
y &= 3*\cos(x)/(2*\cos(x)-1) \\
y &= (\sin(x)-\cos(x))/\sin(x) \\
y &= 2*(\sin(x)) \uparrow 2 - 3*\cos(x) \\
y &= \sin(1/x) \\
y &= (1+\cos(x))/\sin(x)
\end{aligned}$$

Funzioni logaritmiche

$$\begin{aligned}
y &= \log(1+x)/x \\
y &= \log(x)/x \\
y &= (x \uparrow 2 - 2)/(x*\log(x)) \\
y &= \log(\text{abs}((x+1)/(x))) \\
y &= (1+\log(x))/x \\
y &= 2*x - x*\log(\text{abs}(x)) \\
y &= \log(\text{abs}((x+1)/(x-1))) \\
y &= x*\log(\text{abs}(x)) \\
y &= \log(\text{abs}(x \uparrow 2 - x - 2)) \\
y &= \log(\text{abs}((x+3)/(1-x))) \\
y &= x + \log(\text{abs}(2 - \exp(x))) \\
y &= \log(\text{abs}(\sin(x)))
\end{aligned}$$

Funzioni algebriche

$y=(x \uparrow 2-5*x+6)/(6-5*x)$
 $y=(6-5*x)/(x \uparrow 2-5*x+6)$
 $y=(x \uparrow 2+x+1)/(x-1)$
 $y=(-2*x \uparrow 3+3*x \uparrow 2+1)/(x \uparrow y+2*x-1)$
 $y=(-3*x+2)/(x \uparrow 2-3*x+2)$
 $y=(4*x-5)/(x \uparrow 2-1)$
 $y=(x \uparrow 2-2*w+1)/(x \uparrow 2+5*x+4)$
 $y-(x \uparrow 3+1)/(x+2)$
 $y=27/(x \uparrow 2)-8/(x+1) \uparrow 2$
 $y=2*x \uparrow 3/(2*x-1) \uparrow 2$
 $y=x \uparrow 2+1/(1-x \uparrow 2)$
 $y=(x \uparrow 3-3*x+1)/(x \uparrow 2-x)$
 $y=x \uparrow 2+x+1)/(2*x \uparrow 2+3*x+1)$
 $y=x+3/x-1/x \uparrow 2$
 $y=(x \uparrow 3-3*x \uparrow 2)/(4*x \uparrow 2-8)$
 $y=2*x \uparrow 2-abs(5*x+2)$
 $y=(x+3)/abs(x \uparrow 2+x-2)$
 $y=abs(x-1)/(x-1)$
 $y=abs((2*x \uparrow 2-3*x)/(x-2))$
 $y=abs(-2*x \uparrow 3+3*x \uparrow 2+1)/(x \uparrow 3+2*x-1)$
 $y=x \uparrow 3-abs(x)$
 $y=2*x/(1+abs(x))$
 $y=abs(15*x \uparrow 5-x \uparrow 3)$

Funzioni varie

$y=abs(x)*sqr(abs(x))/x$
 $y=sqr(x \uparrow 4-5*x)$
 $y=1/(x+sqr(1+x \uparrow 2))$
 $y=sqr(x \uparrow 2-4*x+3)/(x+1)$
 $y=x+sqr(abs(x \uparrow 2-2*x))$
 $y=x-sqr(abs(x \uparrow 2-2*x))$
 $y=sqr(abs(x))+1/x$
 $y=1/x+sqr(abs(1-x) \uparrow 2)$
 $y=x*sqr(abs(x)/(x-2))$
 $y=(5*x-)/(3*x \uparrow (1/3))$
 $y=(x+sqr(x))/x$
 $y=(x+1)*sin(x)/(x \uparrow 2+2)$
 $y=x \uparrow 2*sin(1/x)$
 $y=x*sin(1/x)$

$$y=(2*x \uparrow 2+3*x-2)*\tan(p*x)$$

$$y=(x \uparrow 2-\sin(x))/(\cos(x)-1)$$

$$y=(x+\sin(x))/x$$

$$y=x \uparrow 2-(\sin(x)) \uparrow 3+\cos(x)$$

$$y=x+2*\sin(x)$$

$$y=x*\cos(x)$$

$$y=1/(x*\cos(x))$$

$$y=\exp(x)*\cos(x)$$

$$y=x*\cos(x)/2-\sin(x)/2$$

$$y=\exp(-x)*\sin(x)$$

$$y=*x-(\tan(x)) \uparrow 2$$

$$y=(x-\exp(\log(x)))/(x-\exp(1))$$

$$y=x-\sin(x)$$

$$y=x*\cos(x)-\sin(x)+p/2$$

$$y=\exp(-x)*(\sin(x)-5*\cos(x))$$

$$y=\log(\text{abs}(1+x))/(2*\sin(x))$$

$$y=\text{abs}(x)*(1+(\sin(1/x)) \uparrow 2)$$

$$y=x-3*\sin(x/2)$$

ESEMPI DI UTILIZZAZIONE

Dalle figure seguenti si può rilevare l'ottimo grado di definizione con cui vengono tracciate le funzioni. Esse rappresentano solo un esempio del valido ausilio didattico che il programma consente di ottenere per studenti di ogni categoria.

Negli ultimi due grafici si può constatare quanto già accennato in precedenza a proposito dell'anomalia legata al comando INT: infatti i due grafici dovrebbero essere simmetrici rispetto all'asse y.




GRAFICO DELLA FUNZIONE :

$$Y = (X^2 - 4X + 5) / (X^2 + 1)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-10
 X (DESTRA)= 8
 Y (INFERIORE)=-1.5
 Y (SUPERIORE)= 8

Fig. 2.2




GRAFICO DELLA FUNZIONE :

$$Y = (X^2 + 1) / (X - X^2)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-8
 X (DESTRA)= 4
 Y (INFERIORE)=-4
 Y (SUPERIORE)= 6

Fig. 2.3




GRAFICO DELLA FUNZIONE :

$$Y = X/(1+X^3)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 5
 Y (INFERIORE)=-2
 Y (SUPERIORE)= 2

Fig. 2.4




GRAFICO DELLA FUNZIONE :

$$Y = ((X-1)^2)*((X-4)^2)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-.5
 X (DESTRA)= 5
 Y (INFERIORE)=-.5
 Y (SUPERIORE)= 10

Fig. 2.5




GRAFICO DELLA FUNZIONE :

$$Y = 3 * \text{SIN}(X) - 50R(3) * \text{COS}(X)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA) = -5
 X (DESTRA) = 15
 Y (INFERIORE) = -10
 Y (SUPERIORE) = 10

Fig. 2.6




GRAFICO DELLA FUNZIONE :

$$Y = 3 * \text{SIN}(2 * X) + 2 * \text{SIN}(4 * X)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA) = -1.5
 X (DESTRA) = 2
 Y (INFERIORE) = -6
 Y (SUPERIORE) = 6

Fig. 2.7




GRAFICO DELLA FUNZIONE :

$$Y = \text{SIN}(2*X) + 2*\text{COS}(X)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 10
 Y (INFERIORE)=-3
 Y (SUPERIORE)= 3

Fig. 2.8




GRAFICO DELLA FUNZIONE :

$$Y = \text{SIN}(X)/X$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-20
 X (DESTRA)= 20
 Y (INFERIORE)=-1
 Y (SUPERIORE)= 2

Fig. 2.9




GRAFICO DELLA FUNZIONE :

$$Y = (X^4 + 2 * X^3) / (X^3 - 1)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA) = -3
 X (DESTRA) = 4
 Y (INFERIORE) = -4
 Y (SUPERIORE) = 8

Fig. 2.10




GRAFICO DELLA FUNZIONE :

$$Y = X^2 * SIN(X) / (X^2 - 1)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA) = -10
 X (DESTRA) = 10
 Y (INFERIORE) = -10
 Y (SUPERIORE) = 10

Fig. 2.11




GRAFICO DELLA FUNZIONE :

$$Y = X^2 - \text{ABS}(5*X+6)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-6
 X (DESTRA)= 8
 Y (INFERIORE)=-15
 Y (SUPERIORE)= 5

Fig. 2.12




GRAFICO DELLA FUNZIONE :

$$Y = \text{ABS}(X) + \text{ABS}(X^2 - X)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-4
 X (DESTRA)= 4
 Y (INFERIORE)=-.5
 Y (SUPERIORE)= 20

Fig. 2.13




GRAFICO DELLA FUNZIONE :

$$Y = \text{SQR}(\text{ABS}((-1+X^3)/(X+2)))$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-10
 X (DESTRA)= 10
 Y (INFERIORE)= 0
 Y (SUPERIORE)= 15

Fig. 2.14




GRAFICO DELLA FUNZIONE :

$$Y = \text{ABS}(X * \text{LOG}(\text{ABS}(X)))$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-2
 X (DESTRA)= 2
 Y (INFERIORE)=-.5
 Y (SUPERIORE)= 2

Fig. 2.15




GRAFICO DELLA FUNZIONE :

$$Y = X/INT(X)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)= 1
 X (DESTRA)= 6
 Y (INFERIORE)=-.5
 Y (SUPERIORE)= 2

Fig. 2.16




GRAFICO DELLA FUNZIONE :

$$Y = EXP((X-1)/(X+1))$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 5
 Y (INFERIORE)=-.5
 Y (SUPERIORE)= 8

Fig. 2.17




GRAFICO DELLA FUNZIONE :

$$Y = \cos(\exp(X))$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-1
 X (DESTRA)= 3.5
 Y (INFERIORE)=-3
 Y (SUPERIORE)= 3

Fig. 2.18




GRAFICO DELLA FUNZIONE :

$$Y = \sin(3*X) + X/2$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 15
 Y (INFERIORE)=-10
 Y (SUPERIORE)= 10

Fig. 2.19




GRAFICO DELLA FUNZIONE :

$$Y = \text{INT}(\text{ABS}(X)) - \text{SQRT}(1 - (X - \text{INT}(X))^2)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 5
 Y (INFERIORE)=-2
 Y (SUPERIORE)= 6

Fig. 2.20




GRAFICO DELLA FUNZIONE :

$$Y = \text{INT}(\text{ABS}(X)) + (X - \text{INT}(X))^2$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 5
 Y (INFERIORE)=-2
 Y (SUPERIORE)= 6

Fig. 2.21

Stampa di funzioni polidrome o sovrapposte con il C64

DESCRIZIONE DEL LISTATO

Con i programmi finora realizzati non è possibile visualizzare funzioni che abbiano più di una ordinata per ciascuna ascissa (per esempio circonferenze, ellissi, iperboli e curve di ordine superiore) oppure più funzioni sovrapposte sullo stesso piano cartesiano.

Pur mantenendo inalterate la struttura di elaborazione, calcolo, tracciamento e stampa, con il seguente programma potremo visualizzare quasi ogni tipo di funzione o più funzioni sovrapposte. L'introduzione delle funzioni non è più realizzata automaticamente, ma deve essere eseguita manualmente inserendo le funzioni stesse nel listato a partire dalla riga 440 fino alla riga 570.

Dopo ogni funzione è necessario aggiungere il comando

:GOSUB 930

per il tracciamento. Infatti nell'intervallo 430-590 si ha un ciclo FOR...NEXT con K che va da 0 a 319, durante il quale per ogni ascissa viene calcolata e tracciata l'ordinata di ciascuna delle funzioni proposte.

La riga 580 apparentemente inutile rappresenta infatti un comodo espediente, come vedremo fra poco. Per battere correttamente tale linea si seguano alla lettera le istruzioni seguenti:

- battere **580 REM "**
- con il tasto **DELETE** cancellare le **"** e ribatterle

- battere otto volte il tasto INSERT (mantenendo premuto lo SHIFT)
- battere altre otto volte il tasto DELETE (senza lo SHIFT)
- battere RUN 200

Se ora provate a chiedere LIST 580 noterete sullo schermo la presenza della scritta RUN 200 senza il relativo numero di riga: basterà andarci sopra con il cursore e premere RETURN per dare il via al programma. Possiamo ora capire che grazie a tale riga, dopo aver introdotto le funzioni nel listato, possiamo far ripartire il programma senza che sia necessario ribattere il comando RUN ed evitando una inutile rilettura dei DATA.

Prima di visualizzare le funzioni è necessario però fissare la scelta degli intervalli.

La stampa si ottiene con il tasto funzione F7 battuto alla fine della visualizzazione.

Ben poco rimane da dire sul resto del programma, molto simile ai precedenti.

```

100 REM *****
110 REM *   COPYRIGHT IN BASIC   *
120 REM *   VIA SEZZE 22 - LATINA *
130 REM *   TEL. 0773 /487631   *
140 REM *-----*
150 REM * STAMPA FUNZIONI POLIVOCHÉ *
160 REM *   0 SOVRAPPOSTE   *
170 REM *****
180 PRINT"[CLEAR][GIALLO]":POKE 53280,14:POKE 53281,6
190 GOSUB 1370
200 PRINT"[CLEAR][GIALLO]"
210 PRINTTAB(5)"[2 DOWN]F1 - F7 VISUALIZ. & STAMPA"
220 PRINTTAB(5)"F3 - CAMBIO DELLE FUNZIONI"
230 PRINTTAB(5)"F5 - SCELTA DEGLI INTERVALLI"
240 GOSUB 850
250 IF QQ<133 OR QQ>135 THEN 240
260 ON QQ-132 GOTO 270,410,610
270 REM -----
280 REM           VISUALIZZAZIONE
290 REM -----
300 SYS 49152
310 DX=(X2-X1)/320:DY=(Y2-Y1)/200
320 GOSUB 980:GOSUB 1090:REM ASSI COORDINATI
330 GOSUB 430
340 GOSUB 850
350 IF QQ=136 THEN GOSUB 780
360 GOSUB 1280
370 GOTO 200
380 REM -----
390 REM           CAMBIO FUNZIONI
400 REM -----
410 PRINT"[HOME][6 DOWN]"
420 LIST 450-580
430 FOR K=0 TO 319
440 X=X1+K*DX:GOSUB 1140
450 Y=SIN(X/2)*COS(X):GOSUB 930
460 Y=1/2*SIN(X)*COS(2*X):GOSUB 930
470 Y=1/3*SIN(1.5*X)*COS(3*X):GOSUB 930

```

```

480 Y=SIN(X/2)+1/2*SIN(X)+1/3*SIN(1.5*X):GOSUB 930
490 :
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :
580 REM "#####RUN 200
590 NEXT K
600 RETURN
610 REM -----
620 REM          CAMBIO INTERVALLI
630 REM -----
640 PRINT"[CLEAR]SCEGLI GLI INTERVALLI DI VISUALIZZAZIONE
650 INPUT "[2 DOWN]X (SINISTRA)=";X1
660 INPUT "X (DESTRA)=";X2
670 IF X1>X2 THEN 640
680 INPUT "[DOWN]Y (BASSA)=";Y1
690 INPUT "Y (ALTA)=";Y2
700 IF Y1>Y2 THEN 680
710 PRINT"[DOWN]CI SONO ASINTOTI VERTICALI ?";
720 GOSUB 850:PRINTQ$:IF Q$="N" THEN NA=-1:GOTO 770
730 PRINT"[DOWN]QUANTI ?";:GOSUB 850:PRINTQ
740 NA=Q:FOR K=1 TO NA
750 PRINT"[DOWN]ASCISSA"K" ASINTOTO =";:INPUT AS(K)
760 NEXT K
770 GOTO 200
780 REM -----
790 REM          STAMPA FUNZIONI
800 REM -----
810 REM ----- TRACCIAM. CORNICE -----
820 FOR K=0 TO 319 STEP 3:Y=0:GOSUB 1210:Y=199:GOSUB 12
10:NEXT K
830 FOR Y=0 TO 199 STEP 3:K=0:GOSUB 1210:K=319:GOSUB 12
10:NEXT Y
840 SYS 52224:RETURN
850 REM -----
860 REM          LOOP DI ATTESA
870 REM -----
880 GET Q$:IF Q$="" THEN 890
890 QQ=ASC(Q$):Q=VAL(Q$):RETURN
900 REM -----
910 REM          ELABORAZIONE COORDINATE
920 REM -----
930 IF YC=Y1 THEN 970
940 IF YC=Y2 THEN 970
950 Y=200-(Y-Y1)/DY
960 GOSUB 1210
970 RETURN
980 REM -----
990 REM          ASSE X
1000 REM -----
1010 IF SGN(Y1)=SGN(Y2) THEN RETURN
1020 Y=INT(Y2/DY+.5)
1030 FOR K=0 TO 319 STEP 6
1040 GOSUB 1210
1050 NEXT K:RETURN
1060 REM -----
1070 REM          ASSE Y
1080 REM -----
1090 IF SGN(X1)=SGN(X2) THEN RETURN
1100 K=-INT(X1/DX+.5)

```

```

1110 FOR Y=0 TO 199 STEP 6
1120 GOSUB 1210
1130 NEXT Y:RETURN
1140 REM -----
1150 REM           ASINTOTI
1160 REM -----
1165 IF NA=-1 THEN RETURN
1170 FOR J=1 TO NA:IF X=AS(J) THEN 1190
1180 NEXT J:RETURN
1190 FOR T=0 TO 199 STEP 6:Y=T:GOSUB 1210:NEXT T
1200 X=X1+(K+1)*DX:RETURN
1210 REM -----
1220 REM           PLOTTAMENTO
1230 REM -----
1240 IF K<0 OR K>319 OR Y<0 OR Y>199 THEN 1270
1250 BY=8192+320*INT(Y/8)+8*INT(K/8)+(Y AND 7)
1260 POKE BY,PEEK(BY) OR 2*(7-(K AND 7))
1270 RETURN
1280 REM -----
1290 REM           DISATTIVAZIONE COMANDO PLOT
1300 REM -----
1310 POKE 53272,PEEK(53272) AND 247
1320 POKE 53265,PEEK(53265) AND 223
1330 PRINT"[CLEAR]":RETURN
1340 REM -----
1350 REM           ROUTINE LETTURA DATA
1360 REM -----
1370 PRINTTAB(10)"[2 DOWN]ATTENDERE PREGO ...."
1380 FOR K= 49152 TO 49222
1390 READ N:POKE K,N:ND=ND+N:NEXT K
1400 POKE 2,2:REM PER STAMPA RISTRETTA SOSTITUIRE CON POKE 2,
1
1410 FOR K=52224 TO 52677
1420 READ N:POKE K,N:ND=ND+N:NEXT K
1430 IF ND<>70923 THEN PRINT"[DOWN]ERRORE NEI DATA":END
1440 RETURN
1450 DATA 173,24,208,9,8,141,24,208
1460 DATA 173,17,208,9,32,141,17,208
1470 DATA 169,0,133,251,169,32,133,252
1480 DATA 160,0,169,0,145,251,200,192
1490 DATA 0,208,249,230,252,169,64,197
1500 DATA 252,208,239,169,0,133,251,169
1510 DATA 4,133,252,160,0,169,3,145
1520 DATA 251,200,192,0,208,249,230,252
1530 DATA 169,8,197,252,208,239,96
1540 REM -----
1550 REM           ROUTINE PER HARD COPY
1560 REM -----
1570 DATA 173,0,221,41,3,73
1580 DATA 3,160,6,10,136,208
1590 DATA 252,141,132,3,173,24
1600 DATA 208,41,8,240,9,24
1610 DATA 169,32,109,132,3,141
1620 DATA 132,3,169,0,32,189
1630 DATA 255,169,4,170,160,255
1640 DATA 32,186,255,32,192,255
1650 DATA 162,4,32,201,255,176
1660 DATA 3,76,61,204,76,32
1670 DATA 205,169,8,32,210,255
1680 DATA 169,13,32,210,255,162
1690 DATA 0,169,1,141,198,205
1700 DATA 169,0,141,199,205,169
1710 DATA 0,141,200,205,169,199
1720 DATA 141,201,205,32,225,255
1730 DATA 208,3,76,32,205,138

```

1740 DATA 72,152,72,32,41,205
1750 DATA 104,168,104,170,173,205
1760 DATA 205,45,203,205,240,12
1770 DATA 173,202,205,13,198,205
1780 DATA 141,202,205,76,143,204
1790 DATA 173,198,205,73,255,45
1800 DATA 202,205,141,202,205,14
1810 DATA 198,205,173,198,205,201
1820 DATA 128,240,20,24,173,199
1830 DATA 205,105,1,141,199,205
1840 DATA 173,200,205,105,0,141
1850 DATA 200,205,76,93,204,173
1860 DATA 202,205,9,128,224,45
1870 DATA 144,10,173,202,205,41
1880 DATA 31,9,128,141,202,205
1890 DATA 142,207,205,166,2,142
1900 DATA 206,205,168,32,210,255
1910 DATA 152,206,206,205,208,246
1920 DATA 174,207,205,169,1,141
1930 DATA 198,205,169,0,141,202
1940 DATA 205,56,173,199,205,233
1950 DATA 6,141,199,205,173,200
1960 DATA 205,233,0,141,200,205
1970 DATA 206,201,205,173,201,205
1980 DATA 201,255,240,3,76,93
1990 DATA 204,224,45,176,31,24
2000 DATA 173,199,205,105,7,141
2010 DATA 199,205,173,200,205,105
2020 DATA 0,141,200,205,232,169
2030 DATA 199,141,201,205,169,13
2040 DATA 32,210,255,76,93,204
2050 DATA 169,13,32,210,255,32
2060 DATA 231,255,96,173,201,205
2070 DATA 41,7,141,204,205,173
2080 DATA 201,205,74,74,74,168
2090 DATA 185,146,205,133,251,185
2100 DATA 172,205,133,252,24,165
2110 DATA 251,109,204,205,133,251
2120 DATA 165,252,105,0,133,252
2130 DATA 24,173,132,3,101,252
2140 DATA 133,252,173,199,205,41
2150 DATA 7,73,7,168,200,169
2160 DATA 0,56,42,136,208,252
2170 DATA 141,203,205,24,173,200
2180 DATA 205,101,252,133,252,173
2190 DATA 199,205,41,248,168,138
2200 DATA 72,120,162,52,134,1
2210 DATA 177,251,162,55,134,1
2220 DATA 88,168,104,170,152,45
2230 DATA 203,205,141,205,205,96
2240 DATA 0,64,128,192,0,64
2250 DATA 128,192,0,64,128,192
2260 DATA 0,64,128,192,0,64
2270 DATA 128,192,0,64,128,192
2280 DATA 0,64,0,1,2,3
2290 DATA 5,6,7,8,10,11
2300 DATA 12,13,15,16,17,18
2310 DATA 20,21,22,23,25,26
2320 DATA 27,28,30,31


ESEMPI DI UTILIZZAZIONE

Nella figura seguente si può notare il tracciamento di una parabola con asse obliquo sovrapposto a una funzione trascendente. La descrizione della parabola

$$x^2 - 2xy + y^2 - x - 2 = 0$$

si realizza spezzandola in due rami corrispondenti alle due funzioni che si ottengono risolvendo l'equazione rispetto alla variabile y:

$$y^2 - 2xy + (x^2 - x - 2) = 0$$



ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-2.5
 X (DESTRA)= 4
 Y (INFERIORE)=-2.5
 Y (SUPERIORE)= 5

```

450 Y=EXP(X)/X+2:GOSUB 930
460 IF X<-2 THEN 590
470 Y=X-SQR(X+2):GOSUB 930
480 Y=X+SQR(X+2):GOSUB 930
490 :
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :
```

Fig. 3.1

Applicando la formula risolutiva ridotta si ha


$$y = x \pm \sqrt{x + 2}$$

Con il segno positivo si avrà il ramo superiore mentre con il segno negativo quello inferiore.

Il tracciamento di questi due rami si verifica alle righe 470 - 480 mentre, per evitare che valori della x minori di -2 rendano negativo il radicando provocando un segnale di errore, nella riga precedente (460) viene imposto un salto condizionato che impedisce il passaggio alle linee sottostanti.

Nella prima riga, la 450, viene definita invece una funzione trascendente che sarà invece tracciata integralmente in quanto si trova a monte del salto condizionato.

Si ricordi di notificare al calcolatore l'esistenza dell'asintoto verticale per $x = 0$.



ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
X (DESTRA)= 5
Y (INFERIORE)=-10
Y (SUPERIORE)= 10

```
450 Y=(X-2)*EXP(X+1):GOSUB 930
460 Y=EXP(1-COS(2*X)):GOSUB 930
470 :
480 :
490 :
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :
```

Fig. 3.2


Nessuna precauzione particolare è invece necessaria per ottenere la Fig. 3.2. La Fig. 3.3 mostra invece un simpatico fiocco ottenuto al solito suddividendo la funzione

$$y^2 = x^3 + x^2$$

nei due rami

$$y = \pm \sqrt{x^3 + x^2}$$

con la solita condizione per evitare il calcolo nei casi in cui il radicando diverrebbe negativo.



ESTREMI DI VISUALIZZAZIONE :


```
X (SINISTRA)=-2
X (DESTRA)= 2
Y (INFERIORE)=-2
Y (SUPERIORE)= 2
```

```
450 Y=X:GOSUB 930
460 Y=-X:GOSUB 930
470 IF X<-1 THEN 590
480 Y=SQR(X^3+X^2):GOSUB 930
490 Y=-SQR(X^3+X^2):GOSUB 930
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :
```

Fig. 3.3

Nelle ultime due figure si ha invece una combinazione di fantasia fra coniche e funzioni trigonometriche.

Si noti che la circonferenza viene riprodotta come un'ellisse perché sui due assi si hanno unità di misura differenti; infatti pur essendo identici gli estremi di visualizzazione orizzontale e verticale, i pixel disponibili per l'asse x sono 320 mentre quelli disponibili per l'asse y sono soltanto 200.




ESTREMI DI VISUALIZZAZIONE :

```
X (SINISTRA)=-1.5
X (DESTRA)= 1.5
Y (INFERIORE)=-1.5
Y (SUPERIORE)= 1.5
```

```
450 IF X<-1 OR X>1 THEN 590
460 Y=SQR(1-X^2):GOSUB 930
470 Y=-SQR(1-X^2):GOSUB 930
480 Y=SIN(2*PI*X)*SQR(1-X^2):GOSUB 930
490 :
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :
```

Fig. 3.4



ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 5
 Y (INFERIORE)=-5
 Y (SUPERIORE)= 5

```

450 Y=-X:GOSUB 930
460 Y=X:GOSUB 930
470 IF X>-1 AND X<1 THEN NEXT
480 Y=SQR(X^2-1):GOSUB 930
490 Y=-SQR(X^2-1):GOSUB 930
500 Y=COS(2*X)*SQR(X^2-1):GOSUB 930
510 :
520 :
530 :
540 :
550 :
560 :
570 :
```

Fig. 3.5

APPLICAZIONI NELLA SCOMPOSIZIONE DI FOURIER

Ogni funzione (anche non periodica) può essere scomposta in un solo modo in una somma di opportune oscillazioni armoniche

$$f(x) = a/2 + a_1 \times \cos(x) + b_1 \times \sin(x) + a_2 \times \cos(2x) + b_2 \times \sin(2x) + \dots$$

dove i coefficienti sono calcolabili con le formule

$$a_k = \frac{1}{\pi} \int_0^{2\pi} f(x) \times \cos(kx) dx$$

$$b_k = \frac{1}{\pi} \int_0^{2\pi} f(x) \times \sin(kx) dx$$

Tale principio è noto come *principio di Fourier*.

Questo teorema è di importanza didattica fondamentale e il programma che stiamo trattando si presta egregiamente alla visualizzazione delle armoniche e della funzione $f(x)$ risultante dalla loro composizione.

Può risultare utile ricordare che:

- se il valore medio di $f(x)$ è nullo allora il termine $a/2$ è assente;
- se $f(x)$ è una curva alternata si hanno solo le armoniche dispari perché solo per esse si può ottenere:

$$f(t + T/2) = -f(t)$$

cioè la curva assume ordinata opposta dopo un semiperiodo;

- se $f(x)$ è simmetrica rispetto all'asse y allora i termini in seno sono nulli perché per tali termini si ha

$$\sin(k) = -\sin(-k)$$

e pertanto per essi non può mai avvenire che

$$f(t) = f(-t)$$

- se $f(x)$ è simmetrica rispetto all'origine deve invece essere verificata la relazione

$$f(t) = -f(-t)$$

e ciò può essere verificato solo per i termini in seno in quanto

$$\cos(k) = \cos(-k)$$

Passiamo all'analisi di alcune funzioni caratteristiche.

Onda triangolare

È composta da infiniti termini dispari in seno di ampiezza decrescente e di segno alternato.

$$y = \sin(x) - 1/9 \times \sin(3x) + 1/25 \times \sin(5x) - \dots$$

Per la sua visualizzazione è sufficiente limitarsi alle prime tre armoniche.

L'intervallo di rappresentazione è

$$\begin{array}{ll} x \text{ (sinistra)} = 0 & x \text{ (destra)} = 6.28 \\ y \text{ (inferiore)} = -2 & y \text{ (superiore)} = 2 \end{array}$$




Fig. 3.6 Onda triangolare.

Onda quadra

$$y = \sin(x) + 1/3 \times \sin(3x) + 1/5 \times \sin(5x) + \dots$$

Estremi di visualizzazione identici ai precedenti.




Fig. 3.7 Onda quadra.

Onda a dente di sega

$$y = \pi/2 - \sin(x) - 1/2 \times \sin(2x) - 1/3 \times \sin(3x) - \dots$$

Intervallo di visualizzazione

$$\begin{array}{ll} x \text{ (sinistra)} = 0 & x \text{ (destra)} = 12.56 \\ y \text{ (inferiore)} = 1.1 & y \text{ (superiore)} = 3.2 \end{array}$$




Fig. 3.8 Onda a dente di sega.

FUNZIONI RECIPROCHE

Una interessante applicazione che proponiamo è costituita dalla possibilità di visualizzare su uno stesso piano cartesiano una generica funzione


$$y = f(x)$$

e la sua corrispondente reciproca

$$y = 1/f(x)$$

Queste due funzioni sovrapposte possono mostrare interessanti relazioni in quanto godono delle seguenti proprietà:

- tracciando le rette orizzontali $y=1$ e $y=-1$ la $f(x)$ e la sua reciproca andranno a intersecarle sempre in punti coincidenti;
- ogni intersezione della $f(x)$ con le ascisse corrisponde a un asintoto verticale per la reciproca e viceversa;
- ogni asintoto orizzontale viene trasformato in un altro asintoto orizzontale con ordinata reciproca;
- ogni punto di massimo si trasforma in un punto di minimo e viceversa (le ordinate saranno ovviamente reciproche).




ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-3
 X (DESTRA)= 6
 Y (INFERIORE)=-6
 Y (SUPERIORE)= 8

```

450 Y=-1:GOSUB 930
460 Y=1:GOSUB 930
470 Y=X^2-5*X+6:GOSUB 930
480 Y=1/(X^2-5*X+6):GOSUB 930
490 :
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :
```

Fig. 3.9



ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 10
 Y (INFERIORE)=-5
 Y (SUPERIORE)= 15

```

450 Y=-1:GOSUB 930
460 Y=1:GOSUB 930
470 Y=(X-2)/(X^2+X-2):GOSUB 930
480 Y=(X^2+X-2)/(X-2):GOSUB 930
490 :
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :
    
```

Fig. 3.10




Fig. 3.11

ESTREMI DI VISUALIZZAZIONE :

```

X (SINISTRA)=-3.5
X (DESTRA)= 3.5
Y (INFERIORE)=-4
Y (SUPERIORE)= 6

```

```

450 Y=-1:GOSUB 930
460 Y=1:GOSUB 930
470 Y=X^4-5*X^2+4:GOSUB 930
480 Y=1/(X^4-5*X^2+4):GOSUB 930
490 :
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :

```

Fig. 3.11 (segue).

SOVRAPPOSIZIONE FRA DUE ONDE: BATTIMENTI

Un ulteriore aspetto, interessante dal punto di vista didattico, è la possibilità di mostrare come due segnali sinusoidali possano combinarsi generando un segnale complesso avente ventri e nodi disposti periodicamente.

Prendiamo in considerazione due funzioni sinusoidali aventi per semplicità la stessa ampiezza a

$$y_1 = a \sin(v_1 x)$$

$$y_2 = a \sin(v_2 x)$$

v_1 e v_2 rappresentano le pulsazioni delle due onde (cioè la loro frequenza moltiplicata per 2π). Sommando le due funzioni si ottiene la nuova funzione

$$y = y_1 + y_2 = a (\sin(v_1 x) + \sin(v_2 x))$$


Applicando la formula di prostaferesi

$$\sin(p) + \sin(q) = 2 \sin((p+q)/2) \cos((p-q)/2)$$

si ha

$$y = 2 a \cos(x(v_1 - v_2)/2) \sin(x(v_1 + v_2)/2)$$

Questa funzione è formata da un fattore in coseno moltiplicato per un secondo fattore in seno; essa è ancora una funzione periodica, ma con ampiezza massima dipendente dal tempo e non costante.



ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-1
 X (DESTRA)= 10
 Y (INFERIORE)=-5
 Y (SUPERIORE)= 5

```

450 Y=COS((5-4)*X)*SIN((5+4)*X):GOSUB 930
460 :
470 :
480 :
490 :
500 :
510 :
520 :
530 :
540 :
550 :
560 :
570 :
```

Fig. 3.12

Il fattore coseno determina l'ampiezza massima dipendente dal tempo mentre il fattore seno genera delle oscillazioni più frequenti la cui ampiezza è fissata in modo variabile dal fattore precedente. La figura mostra chiaramente il segnale sinusoidale modulato in ampiezza.

Si abbiano due moti armonici

$$\begin{aligned}x &= a \sin (v_1 t) \\y &= b \sin (v_2 t + s)\end{aligned}$$

dove v_1 e v_2 rappresentano le pulsazioni, s lo sfasamento tra i due moti e a e b le rispettive ampiezze massime.

Combiniamoli considerandoli come le coordinate parametriche di un punto: in altre parole la figura risultante dalla combinazione avrà ascissa e ordinata coincidenti con i due moti armonici. Il parametro t è il tempo. Su alcuni testi di fisica sono spesso riportate tali figure, denominate figure di Lissajou, e ottenute per la prima volta con il doppio pendolo del Righi. Lo svolgimento di questo argomento può utilmente essere sfruttato per applicazioni didattiche durante la trattazione dei moti armonici.

Il luogo geometrico descritto dal punto avente le suddette coordinate parametriche può ottenersi eliminando il parametro t , ricavandolo dalla prima funzione e sostituendolo nella seconda.

Cominciamo ad analizzare il caso in cui $v_1 = v_2$, ricordando che

$$\cos (v_1 t) = \sin \sqrt{(1 - (\sin (v_1 t))^2)}$$

Eseguendo i calcoli si ricava l'equazione di una ellisse ruotata con centro nell'origine: infatti la presenza del termine in xy denuncia la rotazione, mentre la mancanza dei termini lineari in x e y evidenzia il fatto che il centro di simmetria coincide con l'origine degli assi.

$$\frac{x^2}{a^2} + \left(\frac{\frac{y}{b} - \frac{x}{a} \cos (s)}{\sin (s)} \right)^2 = 1$$

Nel caso in cui $a = b$ si ha una circonferenza.

È interessante la trattazione della evoluzione della figura quando lo sfasamento s varia fra 0 e 180 gradi.

Per $s = 0$ si ha $y = (b/a)x$, cioè un segmento rettilineo con centro nell'origine degli assi e coefficiente angolare b/a .

Per valori compresi fra 0 e 90 gradi si ottiene una ellisse con centro nell'origine e asse maggiore inclinato verso l'alto.

Per $s = 90$ gradi l'ellisse è orizzontale con equazione canonica

$$\frac{y^2}{b^2} + \frac{x^2}{a^2} = 1$$

Per valori compresi fra 90 e 180 gradi si ottiene di nuovo una ellisse, ma ruotata verso il basso.

Infine per $s = 180$ gradi si ha un segmento rettilineo di equazione

$$y = (-b/a)x.$$

Se invece v_1 e v_2 non sono uguali si ottiene una figura variamente intrecciata.

```

10 REM *****
20 REM *          TRACCIAMENTO          *
30 REM *          FIGURE DI LISSAJOU   *
40 REM *****
50 CLR:POKE 53280,1:POKE 53281,6
60 PRINT"[CLEAR]":GOSUB 1340
70 PRINT"[CLEAR][DOWN]TAB(10)"FIGURE DI LISSAJOU"
80 INPUT "[DOWN]PULSAZIONE ORIZZONTALE";PX
85 INPUT "[DOWN]PULSAZIONE VERTICALE";PY
90 INPUT "[DOWN]SFASAM. TRA I DUE MOTI (IN GRADI)";S
100 SYS 49152
110 K=0:S=S*PI/180
120 X=160*SIN(K*PX)+160
130 Y=200-100*SIN(K*PY+S)+100
200 GOSUB 1240
500 K=K+.05:IF K*PX<2*PI OR K*PY+S<2*PI THEN 120
510 GOSUB 800
520 IF Q$="" THEN GOSUB 820
530 GOSUB 1310
540 PRINT"[CLEAR]ANCORA ?":GOSUB 800
550 IF Q$="S" THEN 70
640 PRINT"[2 DOWN]OK - FINE LAVORO." :END
780 REM -----
790 REM          STAMPA FIGURA
800 REM -----
810 REM ----- TRACCIAM. CORNICE -----
820 FOR K=0 TO 319 STEP 3:Y=0:GOSUB 1210:Y=199:GOSUB 12
10:NEXT K
830 FOR Y=0 TO 199 STEP 3:K=0:GOSUB 1210:K=319:GOSUB 12
10:NEXT Y
840 SYS 52224:RETURN
850 REM -----
860 REM          LOOP DI ATTESA
870 REM -----
880 GET Q$:IF Q$="" THEN 800
890 QQ=ASC(Q$):Q=VAL(Q$):RETURN
1210 REM -----
1220 REM          PLOTTAMENTO
1230 REM -----
1240 IF X<0 OR X>319 OR Y<0 OR Y>199 THEN 1270
1250 BY=8192+320*INT(Y/8)+8*INT(X/8)+(Y AND 7)
1260 POKE BY,PEEK(BY) OR 2*(7-(X AND 7))
1270 RETURN
1280 REM -----
1290 REM          DISATTIVAZIONE COMANDO PLOT
1300 REM -----

```

```

1310 POKE 53272,PEEK(53272) AND 247
1320 POKE 53265,PEEK(53265) AND 223
1330 PRINT"[CLEAR]":RETURN
1340 REM -----
1350 REM      ROUTINE LETTURA DATA
1360 REM -----
1370 PRINTTAB(10)"[2 DOWN]ATTENDERE PREGO ...."
1380 FOR K= 49152 TO 49222
1390 READ N:POKE K,N:ND=ND+N:NEXT K
1400 POKE 2,2:REM PER STAMPA RISTRETTA SOSTITUIRE CON POKE 2,
1
1410 FOR K=52224 TO 52677
1420 READ N:POKE K,N:ND=ND+N:NEXT K
1430 IF ND<>70923 THEN PRINT"[DOWN]ERRORE NEI DATA":END
1440 RETURN
1450 DATA 173,24,208,9,8,141,24,208
1460 DATA 173,17,208,9,32,141,17,208
1470 DATA 169,0,133,251,169,32,133,252
1480 DATA 160,0,169,0,145,251,200,192
1490 DATA 0,208,249,230,252,169,64,197
1500 DATA 252,208,239,169,0,133,251,169
1510 DATA 4,133,252,160,0,169,3,145
1520 DATA 251,200,192,0,208,249,230,252
1530 DATA 169,8,197,252,208,239,96
1540 REM -----
1550 REM      ROUTINE PER HARD COPY
1560 REM -----
1570 DATA 173,0,221,41,3,73
1580 DATA 3,160,6,10,136,208
1590 DATA 252,141,132,3,173,24
1600 DATA 208,41,8,240,9,24
1610 DATA 169,32,109,132,3,141
1620 DATA 132,3,169,0,32,189
1630 DATA 255,169,4,170,160,255
1640 DATA 32,186,255,32,192,255
1650 DATA 162,4,32,201,255,176
1660 DATA 3,76,61,204,76,32
1670 DATA 205,169,8,32,210,255
1680 DATA 169,13,32,210,255,162
1690 DATA 0,169,1,141,198,205
1700 DATA 169,0,141,199,205,169
1710 DATA 0,141,200,205,169,199
1720 DATA 141,201,205,32,225,255
1730 DATA 208,3,76,32,205,138
1740 DATA 72,152,72,32,41,205
1750 DATA 104,168,104,170,173,205
1760 DATA 205,45,203,205,240,12
1770 DATA 173,202,205,13,198,205
1780 DATA 141,202,205,76,143,204
1790 DATA 173,198,205,73,255,45
1800 DATA 202,205,141,202,205,14
1810 DATA 198,205,173,198,205,201
1820 DATA 128,240,20,24,173,199
1830 DATA 205,105,1,141,199,205
1840 DATA 173,200,205,105,0,141
1850 DATA 200,205,76,93,204,173
1860 DATA 202,205,9,128,224,45
1870 DATA 144,10,173,202,205,41
1880 DATA 31,9,128,141,202,205
1890 DATA 142,207,205,166,2,142
1900 DATA 206,205,169,32,210,255
1910 DATA 152,206,206,205,208,246
1920 DATA 174,207,205,169,1,141
1930 DATA 198,205,169,0,141,202
1940 DATA 205,56,173,199,205,233

```

```

1950 DATA 6,141,199,205,173,200
1960 DATA 205,233,0,141,200,205
1970 DATA 206,201,205,173,201,205
1980 DATA 201,255,240,3,76,93
1990 DATA 204,224,45,176,31,24
2000 DATA 173,199,205,105,7,141
2010 DATA 199,205,173,200,205,105
2020 DATA 0,141,200,205,232,169
2030 DATA 199,141,201,205,169,13
2040 DATA 32,210,255,76,93,204
2050 DATA 169,13,32,210,255,32
2060 DATA 231,255,96,173,201,205
2070 DATA 41,7,141,204,205,173
2080 DATA 201,205,74,74,74,168
2090 DATA 185,146,205,133,251,185
2100 DATA 172,205,133,252,24,165
2110 DATA 251,109,204,205,133,251
2120 DATA 165,252,105,0,133,252
2130 DATA 24,173,132,3,101,252
2140 DATA 133,252,173,199,205,41
2150 DATA 7,73,7,168,200,169
2160 DATA 0,56,42,136,208,252
2170 DATA 141,203,205,24,173,200
2180 DATA 205,101,252,133,252,173
2190 DATA 199,205,41,248,168,138
2200 DATA 72,120,162,52,134,1
2210 DATA 177,251,162,55,134,1
2220 DATA 88,168,104,170,152,45
2230 DATA 203,205,141,205,205,96
2240 DATA 0,64,128,192,0,64
2250 DATA 128,192,0,64,128,192
2260 DATA 0,64,128,192,0,64
2270 DATA 128,192,0,64,128,192
2280 DATA 0,64,0,1,2,3
2290 DATA 5,6,7,8,10,11
2300 DATA 12,13,15,16,17,18
2310 DATA 20,21,22,23,25,26
2320 DATA 27,28,30,31

```

Le pulsazioni v_1 e v_2 corrispondono nel listato alle variabili PX e PY, le ampiezze a e b corrispondono rispettivamente ai valori 160 e 100 (in modo che la figura risultante venga a tutto schermo).

Le due funzioni sono introdotte nelle righe 120 e 130 e si noti come è necessario aggiungere alle coordinate i valori 100 e 160 per trasportare l'origine degli assi nel punto centrale dello schermo. Inoltre per le ordinate è necessario determinare il complemento a 200 per ovviare al fatto che l'asse y dello schermo è orientato verso il basso, mentre nel piano cartesiano esso è normalmente orientato verso l'alto.

Il parametro t nel nostro caso è rappresentato dalla variabile K che, nella riga 500, viene successivamente incrementata fino a quando almeno una delle due componenti non abbia eseguito un ciclo completo.

Poiché l'argomento del seno deve essere espresso in radianti, nella riga 110 lo sfasamento viene trasformato da gradi in radianti.

Per il resto il programma segue fedelmente i criteri già visti.




Fig. 3.13

Nella prima figura abbiamo posto $PX = PY = 1$, $S = 45$ gradi, ottenendo così un'ellisse ruotata verso l'alto come previsto teoricamente. Nella seconda figura invece è stato posto $PX = 1$, $PY = .5$, $S = 0$. Infine nella terza figura invece è stato posto $PX = 1$, $PY = 3$, $S = 90$, ed è stato necessario trasformare l'incremento $.05$ alla riga 500 in un incremento più piccolo $.01$, altrimenti i punti della figura sarebbero risultati troppo distanziati fra loro.




Fig. 3.14




Fig. 3.15

Stampa di funzioni con il Simon's Basic

DESCRIZIONE DEL LISTATO

Ci occuperemo ora di un programma che richiede l'impiego del Simon's Basic, un importante strumento che aggiunge molti nuovi comandi (oltre 100) al Basic standard del Commodore 64.

È un programma in linguaggio macchina (facilmente reperibile, ma che non possiamo fornire in questo libro perché coperto da Copyright) che va caricato e al quale va dato il RUN.

Da quel momento il Commodore 64 è in grado, con opportuni e semplici comandi Basic, di operare in alta risoluzione, di avere una gestione facilitata del suono, di trasmettere su stampante il contenuto dello schermo, di operare con programmazione strutturata ecc.

Noi ci limiteremo a usare soltanto alcuni comandi di tracciamento e l'*hardcopy*, che ci eviteranno il ricorso alle routine in DATA e inoltre consentiranno una esecuzione dei grafici molto più veloce.

In più sarà possibile scrivere con caratteri grafici sullo schermo in alta risoluzione.

```
100 REM *****
110 REM *   COPYRIGHT IN BASIC   *
120 REM *   VIA SEZZE 22 - LATINA *
130 REM *   TEL. 0773 /427631    *
140 REM *-----*
150 REM * STAMPA FUNZIONI POLIVOCHE *
160 REM *       0 SOVRAPPOSTE      *
170 REM *****
```



```

842 PRINT#4:PRINT#4,SPC(25)"X (SINISTRA)=""X1
843 PRINT#4,SPC(25)"X (DESTRA)=""X2
844 PRINT#4,SPC(25)"Y (INFERIORE)=""Y1
845 PRINT#4,SPC(25)"Y (SUPERIORE)=""Y2
846 PRINT#4:CMD4:LIST 450-570:PRINT#4:END
850 REM -----
860 REM          LOOP DI ATTESA
870 REM -----
880 GET Q$:IF Q$="" THEN 880
890 QQ=ASC(Q$):Q=VAL(Q$):RETURN
900 REM -----
910 REM          ELABORAZIONE COORDINATE
920 REM -----
930 IF Y<=Y1 THEN 970
940 IF Y>=Y2 THEN 970
950 Y=200-(Y-Y1)/DY
960 PLOT K,Y,1
970 RETURN
980 REM -----
990 REM          ASSE X
1000 REM -----
1010 IF SGN(Y1)=SGN(Y2) THEN RETURN
1020 Y=INT(Y2/DY+.5)
1025 CHAR 305,Y+5,24,1,1
1030 FOR K=0 TO 319 STEP 3
1040 PLOT K,Y,1
1050 NEXT K:RETURN
1060 REM -----
1070 REM          ASSE Y
1080 REM -----
1090 IF SGN(X1)=SGN(X2) THEN RETURN
1100 K=-INT(X1/DX+.5)
1105 CHAR K+5,5,25,1,1
1110 FOR Y=0 TO 199 STEP 3
1120 PLOT K,Y,1
1130 NEXT Y:RETURN
1140 REM -----
1150 REM          ASINTOTI
1160 REM -----
1165 IF NA=-1 THEN RETURN
1170 FOR J=1 TO NA:IF X>AS(J)-.001 AND X<AS(J)+.001 THEN 1190
1180 NEXT J:RETURN
1190 FOR Y=0 TO 199 STEP 6:PLOT K,Y,1:NEXT Y
1200 X=X1+(K+1)*DX:RETURN

```

L'alta risoluzione si ottiene nella riga 300 con il solo comando:

HIRES 6,3

Il primo numero stabilisce il colore della cornice del video mentre il secondo quello della parte centrale dello schermo.

I colori possono essere variati a piacere usando il codice seguente leggermente differente da quello del Basic standard.

0 = nero
2 = rosso

1 = bianco
3 = cyan

4 = porpora	5 = verde
6 = blu	7 = giallo
8 = arancione	9 = arancio chiaro
10 = rosa	11 = cyan chiaro
12 = porpora chiaro	13 = verde chiaro
14 = celeste	15 = giallo chiaro

Il ritorno alla bassa risoluzione si ottiene con il comando **NRM** nella riga 370.

Il tracciamento si ottiene con il comando:

PLOT a,b,c

I primi due numeri rappresentano le coordinate del pixel interessato, mentre l'ultima fissa le modalità di tracciamento cioè:

C = 0 il pixel viene spento

C = 1 il pixel viene acceso

C = 2 il pixel viene acceso se spento o viceversa

Le righe contenenti tale comando sono: 820, 838, 160, 1040, 1120, 1190.

Il trasferimento del grafico dal video sulla stampante si ottiene semplicemente con il comando **COPY** alla riga 840.

Nelle righe seguenti vengono scritti gli estremi di visualizzazione e il listato delle linee contenenti le funzioni graficate.

Relativamente al tracciamento degli asintoti può accadere che pur senza verificarsi un errore causato da una divisione per zero il programma si interrompa perché l'ordinata diviene talmente grande da generare un errore di *overflow*.

Questo raro inconveniente può essere il più delle volte evitato imponendo il tracciamento dell'asintoto verticale non appena l'ascissa (assoluta) della funzione cada entro un opportuno intervallo contenente l'ascissa dell'asintoto.

Più precisamente, chiamando con **XA** l'ascissa di un asintoto verrà forzato il tracciamento non appena risulterà

$$XA - .001 < X < XA + .001$$

A questo pensa la riga 1170.

La struttura del resto del programma è molto simile a quella realizzata nel primo capitolo.

RICERCA DEI MASSIMI E MINIMI DI UNA FUNZIONE

Accingiamoci ora a realizzare un programma che consenta la ricerca automatica dei punti caratteristici di una funzione (massimi, minimi e flessi orizzontali) e alla elaborazione di una tabella contenente le ascisse, le ordinate, le derivate prime e seconde di un insieme di punti discreti contenuti in un certo intervallo.

```

100 REM *****
110 REM *
120 REM * STUDIO DI UNA FUNZIONE *
130 REM *
140 REM *****
150 CLR :PRINT "[CLEAR][BLEU]":POKE 53281,15:POKE 53280,8
160 PRINT "[CLEAR]BATTI LA FUNZIONE : "
170 INPUT "[3 DOWN]Y = ";Y$
180 PRINT "[CLEAR][3 DOWN]220 DEF FNY(X)="Y$
185 PRINT "225 Y$="CHR$(34)Y$CHR$(34)
190 PRINT "GOTO 220[HOME]"
200 FOR K=0 TO 2:POKE 631+K,13:NEXT:POKE 198,3
210 END
220 DEF FNY(X)=-X15/5-X14/4+X13/3+X12/2-3
225 Y$="-X15/5-X14/4+X13/3+X12/2-3"
230 DEF FNY1(X)=(FNY(X+1E-4)-FNY(X-1E-4))/2E-4
240 DEF FNY2(X)=(FNY1(X+1E-4)-FNY1(X-1E-4))/2E-4
250 DEF FNAR(X)=INT(X*1E2+.5)/1E2
260 POKE 198,0
270 PRINT "[CLEAR] OK - ORA DIMMI ENTRO QUALE INTERVALLO"
280 PRINT " VUOI ESAMINARLA : "
290 INPUT "[I DOWN] X (SINISTRA)=";X1
300 INPUT " X (DESTRA)=";X2
310 IF X1>X2 THEN 270
320 REM -----
330 REM RICERCA MAX MIN & FLEX
340 REM -----
350 PRINTTAB(9)"[ROSSO][2 DOWN]UN ATTIMO DI PAZIENZA[BLEU]"
400 DIM M$(20):H=1:I=SGN(FNY1(X1)):S=.25:XX=X1
500 IF SGN(FNY1(XX))<>I THEN S=-S/10:I=SGN(FNY1(XX))
510 IF FNY1(XX)=0 THEN 560
520 XX=XX+S:IF XX<X2 THEN 500
530 GOTO 630:REM USCITA PER FINE
560 M=FNAR(XX)
570 IF SGN(FNY2(XX-.1))<>SGN(FNY2(XX+.1)) THEN 590
575 IF FNY(XX+.1)<FNY(XX) THEN 600
580 IF FNY(XX+.1)>FNY(XX) THEN 610
585 GOTO 520
590 W=FNAR(FNY(XX)):M$(H)="FLESSO = (" +STR$(M)+";"+STR$(W)+")"
"
595 PRINT:PRINTM$(H):PRINT:GOTO 620
600 W=FNAR(FNY(XX)):M$(H)="MASSIMO = (" +STR$(M)+";"+STR$(W)+")"
)"
605 PRINT:PRINTM$(H):PRINT:GOTO 620
610 W=FNAR(FNY(XX)):M$(H)="MINIMO = (" +STR$(M)+";"+STR$(W)+")"
"
615 PRINT:PRINTM$(H):PRINT
620 H=H+1:I=SGN(FNY1(XX+.1)):S=.25:GOTO 500
630 PRINT "[I DOWN] VUOI UN ELENCO CON LE ORDINATE E LE"
640 PRINT "DERIVATE PRIME E SECONDE ?"
650 GOSUB 1000

```

```

670 IF Q$<>"S" THEN 900
680 PRINT"[DOWN]VUOI ANCHE LA STAMPA ?":GOSUB 1000
690 IF Q$<>"S" THEN 730
695 GOSUB 2030
700 REM -----
710 REM          E L E N C O
720 REM -----
730 T$="[ROSSO] X F(X) F'(X) F''(X)"
740 PRINT"[CLEAR][DOWN]T$"
750 FOR K=1 TO 40:PRINT"[ARANC]_":NEXT
760 PRINT"[BLEU][DOWN]":IF FL=1 THEN GOSUB 2040
770 FOR X=X1 TO X2 STEP .25
780 PRINTX,FNAR(FNY(X)),FNAR(FNY1(X)),FNAR(FNY2(X))
790 IF FL=1 THEN GOSUB 2050
800 T=T+1:IF INT(T/16)<>T/16 THEN 850
810 PRINTTAB(12)"[DOWN](PREMI UN TASTO)":GOSUB 1000
820 PRINT"[CLEAR][DOWN]T$"
830 FOR K=1 TO 40:PRINT"[ARANC]_":NEXT
840 PRINT"[BLEU][DOWN]"
850 NEXT
855 IF FL=1 THEN GOSUB 2130
860 PRINT"[DOWN]ANCORA ?":GOSUB 1000
870 IF Q$="S" THEN RUN
900 PRINT"[DOWN]OK - FINE LAVORO.":END
1000 REM -----
1010 REM          C I C L O  D I  A T T E S A
1020 REM -----
1030 GET Q$:IF Q$="" THEN 1030
1040 RETURN
2000 REM -----
2010 REM          S T A M P A
2020 REM -----
2030 OPEN 4,4:PRINT#4,CHR$(14):FL=1:RETURN
2040 PRINT#4,T$:PRINT#4:FOR K=1 TO 40:PRINT#4,"#":NEXT
2045 PRINT#4:PRINT#4:RETURN
2050 PRINT#4,X)
2060 PRINT#4,CHR$(16)"20"FNAR(FNY(X));
2070 PRINT#4,CHR$(16)"40"FNAR(FNY1(X));
2080 PRINT#4,CHR$(16)"60"FNAR(FNY2(X))
2090 RETURN
2100 REM -----
2110 REM          S T A M P A  F I N A L E
2120 REM -----
2130 PRINT#4:FOR K=1 TO 40:PRINT#4,"#":NEXT
2140 PRINT#4:PRINT#4:PRINT#4,SPC(10)"Y ="Y$
2150 PRINT#4:FOR K=1 TO H
2160 PRINT#4,M$(K)
2170 NEXT K
2180 RETURN

```

Nella introduzione della funzione appare una nuova riga (la 185). Quando il programma riparte dalla riga 220 (dopo i tre RETURN automatici) le variabili vengono tutte azzerate e quindi la Y\$ battuta nella riga 170 viene "dimenticata" rendendo impossibile la stampa su carta del testo della funzione dopo il grafico. In 185 viene quindi forzato l'inserimento della nuova riga 225 contenente la funzione sotto forma di stringa che non sarà più dimenticata in

quanto si trova dopo il punto in cui il programma ricomincia a scorrere (riga 220).

Nelle righe 230 - 240 sono definite due nuove funzioni, le quali non sono altro che i rapporti incrementali della funzione proposta.

$$y'(x) = \frac{y(x + 10^{-4}) - y(x - 10^{-4})}{2 \times 10^{-4}}$$

$$y''(x) = \frac{y'(x + 10^{-4}) - y'(x - 10^{-4})}{2 \times 10^{-4}}$$

Essendo l'incremento sufficientemente piccolo, con buona approssimazione possiamo confondere i rapporti incrementali con le derivate prima e seconda calcolate nel punto x .

Nella riga 250 si ha una terza funzione con il solo scopo di arrotondare i risultati ottenuti prendendo in considerazione solo due cifre decimali. Per la ricerca dei massimi, minimi e flessi la routine interessata va dalla riga 490 alla 620 e analizza un intervallo ($X1-X2$) scelto precedentemente.

In questo intervallo facciamo scorrere un punto da $X1$ a $X2$ con incrementi successivi di lunghezza S . Indichiamo con XX tale ascissa variabile e prendiamo in considerazione il segno D della derivata prima calcolata in XX .

Fino a quando tale segno rimane invariato si avrà la funzione sempre crescente (se $D > 0$) o decrescente (se $D < 0$). Non appena verrà registrato un cambiamento di segno, avremo in corrispondenza un punto di minimo, di massimo o di flesso orizzontale.

Per determinare con maggiore precisione le ascisse di tale punto, non appena D si inverte cambiamo segno all'incremento S rendendolo inoltre più piccolo: l'ascissa XX tornerà indietro più "lentamente" e comincerà a oscillare avanti e indietro fino a quando non venga determinato esattamente il punto in cui la derivata si annulla.

Tale situazione viene rilevata nella riga 510 dove viene effettuato un salto in 560 per l'arrotondamento e il riconoscimento (nelle linee successive) delle caratteristiche del punto (massimo, minimo o flesso).

Al termine di tale riconoscimento, nella riga 620 si ha un ritorno indietro per il proseguimento dell'analisi del rimanente intervallo.

Tale ricerca prosegue fino a quando XX non risulta minore di $X2$, e al verificarsi di tale circostanza si ha in 530 il salto per uscire dalla routine.

Il riconoscimento dei punti caratteristici avviene nelle righe 570 - 615 semplicemente confrontando le ordinate della funzione e i segni della derivata seconda.

Sia x_0 l'ascissa di uno di tali punti caratteristici: se il segno della derivata

seconda in $(x_0 - 0.1)$ è differente dal segno calcolato in $(x_0 + 0.1)$ allora la funzione ha in x_0 una inversione di concavità e quindi x_0 rappresenta l'ascissa di un flesso orizzontale.

Se invece

$$f(x_0) \geq f(x_0 + 0.1)$$

allora x_0 è rispettivamente un massimo o un minimo con coordinata

$$x_0 f(x_0)$$

Il contatore H (riga 620) serve per la memorizzazione dei punti caratteristici via via rilevati nella variabile M\$(H) dimensionata nella riga 490. Il resto del listato non presenta particolari difficoltà di comprensione in quanto elabora semplicemente una tabella con ascisse, ordinate, derivate prime e seconde nell'intervallo da X1 a X2 con incremento .25. Alla riga 2000 si ha la stampa della tabella sviluppata per la stampante MPS/801, ma facilmente modificabile per la MPS/802 o altre stampanti eliminando il CHR\$(16) nelle righe 2060 - 2080 che serve (unitamente alle due cifre tra virgolette) alla tabulazione.

Nelle righe finali si ha la stampa dei punti caratteristici precedentemente calcolati.

GRAFICO DI UNA FUNZIONE E CALCOLO DI UN'AREA

Mentre il programma precedente non richiede l'uso del Simon's Basic, esso è necessario per quello che stiamo per esaminare. Una novità è costituita dalla possibilità di calcolare un integrale definito e di vedere la corrispondente area sul grafico colorarsi gradualmente dopo il calcolo. La funzione da graficare viene introdotta col solito criterio di autostampa: vengono fissati gli estremi di visualizzazione, forniti gli eventuali asintoti e stabilito se si vuole calcolare un'area.

In caso affermativo viene attivato un *flag* in 325 e si fissano gli estremi di integrazione AI e AF.

La struttura del programma è sostanzialmente identica a quella già vista nel capitolo secondo.

L'unica differenza riguarda il calcolo dell'area, ottenuto senza particolari algoritmi (Gauss-Seidel ecc...), e la colorazione sullo schermo della regione interessata.

```

100 CLR:PRINT" "
110 PRINT"CBATTI LA FUNZIONE : "
120 INPUT"Y = ";Y$
130 PRINT"DEF FNY(X)="Y$
140 PRINT"Y$="CHR$(34)Y$CHR$(34)
150 PRINT"GOTO 180"
160 FOR K=0 TO 3:POKE 631+K,13:NEXT:POKE 198,4
170 END
180 DEF FNY(X)=SIN(X)
190 Y$="SIN(X)"
195 DEF FNAR(Y)=INT(Y*1E2+.5)/1E2
200 POKE 198,0:PRINT"OK - OKA "
210 PRINT"SCEGLI GLI INTERVALLI DI "
220 PRINT TAB(5)"VISUALIZZAZIONE : "
230 INPUT"X (SINISTRA)=";X1
240 INPUT"X (DESTRA)=";X2
250 IF X1>X2 THEN 210
260 INPUT"Y (BASSA)=";Y1
270 INPUT"Y (ALTA)=";Y2
280 IF Y1>Y2 THEN 210
290 PRINT"CI SONO ASINTOTI VERTICALI ? ";
300 GOSUB 1080:PRINT Q$:IF Q$="N" THEN NA=-1:GOTO 322
310 PRINT"QUANTI ? ";GOSUB 1080:PRINT Q$:NA=Q
320 PRINT"FOR K=1 TO NA:PRINT"ASCISSA"K"" ASINTOTO =";INPUT
AS(K):NEXT K
322 PRINT"VUOI CALCOLARE UN' AREA ? ";GOSUB 1080:PRINT Q$
324 IF Q$<>"S" THEN 360
325 FL=1:REM FLAG PER CALCOLO AREA
326 INPUT"ASCISSA INIZIALE";AI
327 INPUT"ASCISSA FINALE";AF
330 REM -----
340 REM          INIZIO ELABORAZIONE
350 REM -----
360 HIRES 6,3
370 IX=(X2-X1)/320:DY=(Y2-Y1)/200
380 GOSUB 570:GOSUB 660
390 FOR X=0 TO 319
400 X0=X1+X*IX:GOSUB 960
410 Y0=FNY(X0)
420 IF Y0<=Y1 THEN 460
430 IF Y0>=Y2 THEN 460
440 Y=200-(Y0-Y1)/DY
450 PLOT X,Y,1
460 NEXT X
470 GOSUB 1130:GOSUB 750
480 PRINT"          VUOI CAMBIARE GLI INTERVALLI DI "
490 PRINT"VISUALIZZAZIONE ? "
500 GOSUB 1080
510 IF Q$="S" THEN PRINT" " " ";GOTO 210
520 PRINT"VUOI CAMBIARE LA FUNZIONE ? "
530 GOSUB 1080
540 IF Q$="S" THEN RUN
550 PRINT"OK - FINE LAVORO."
560 END
570 REM -----
580 REM          ASSE X
590 REM -----
600 IF SGN(Y1)=SGN(Y2) THEN RETURN
610 Y=INT(Y2/DY+.5)
620 CHAR 305,Y+5,24,1,1
630 FOR X=0 TO 319
640 PLOT X,Y,1
650 NEXT X:RETURN
660 REM -----
670 REM          ASSE Y

```

```

680 REM -----
690 IF SGN(X1)=SGN(X2) THEN RETURN
700 X=-INT(X1/IX+.5)
710 CHAR X+5.5,25.1,1
720 FOR Y=0 TO 199
730 PLOT X,Y,1
740 NEXT Y:RETURN
750 REM -----
760 REM RITORNO IN BASSA RISOLUZIONE
770 REM -----
780 GOSUB 1000
790 IF Q$="" THEN GOSUB 810
800 NRM:PRINT "D":RETURN
810 REM -----
820 REM          STAMPA
830 REM -----
840 FOR X=0 TO 319:PLOT X,0,1:PLOT X,199,1:NEXT
850 FOR Y=0 TO 199:PLOT 0,Y,1:PLOT 319,Y,1:NEXT
860 COPY
870 OPEN 4,4
880 PRINT#4:PRINT#4:PRINT#4,SPC(5)"GRAFICO DELLA FUNZIONE : "
890 PRINT#4:PRINT#4,SPC(30)"Y = "Y$
900 PRINT#4:PRINT#4,SPC(5)"ESTREMI DI VISUALIZZAZIONE : "
910 PRINT#4:PRINT#4,SPC(25)"X (SINISTRA)="X1
920 PRINT#4,SPC(25)"X (DESTRA)="X2
930 PRINT#4,SPC(25)"Y (INFERIORE)="Y1
940 PRINT#4,SPC(25)"Y (SUPERIORE)="Y2
942 PRINT#4:PRINT#4,SPC(5)"ESTREMI DI INTEGRAZIONE : "
944 PRINT#4:PRINT#4,SPC(25)"X1 ="AI
946 PRINT#4,SPC(25)"X2 ="AF
950 CLOSE 4:RETURN
960 REM -----
970 REM          ASINTOTI VERTICALI
980 REM -----
990 IF NA=-1 THEN 1070
1000 FOR J=1 TO NA
1010 IF X0=AS(J) THEN 1030
1020 NEXT J:RETURN
1030 FOR Y=0 TO 199 STEP 2
1040 PLOT X,Y,1
1050 NEXT Y
1060 X=X+1:X0=X1+X*DX
1070 RETURN
1080 REM -----
1090 REM          CICLO DI ATTESA
1100 REM -----
1110 GET Q$:IF Q$="" THEN 1110
1120 Q=VAL(Q$):RETURN
1130 REM -----
1140 REM          INTEGRAZIONE
1150 REM -----
1160 IF FL=0 THEN RETURN
1170 SU=(AF-AI)/200
1180 FOR K=AI TO AF-SU STEP SU
1190 IN=IN+(FNY(K)+FNY(K+SU))*SU/2
1200 NEXT K
1210 R$=STR$(FNAR(IN)):R$="AREA="+R$
1220 TEXT 5.5,R$,1,1,8
1230 FOR X0=AI TO AF STEP DX
1240 X=(X0-X1)/IX
1250 AY=INT(Y2/DY+.5)
1255 YY=200-(FNY(X0)-Y1)/DY
1260 LINE X,AY,X,YY,1
1270 NEXT X0
1300 FL=0:RETURN

```

La routine interessata al calcolo dell'area è quella che va dalla riga 1160 alla fine.

In 1160 viene forzato un RETURN nel caso in cui sia nullo il flag relativo all'opzione di calcolo dell'area.

L'intervallo compreso fra gli estremi di integrazione AI e AF viene ripartito in 200 parti uguali (SU) e quindi in un ciclo FOR...NEXT viene realizzato il calcolo dello scaloide che approssima accettabilmente l'area della regione curvilinea.




Fig. 4.1

Grazie alla suddivisione effettuata, possiamo scomporre la regione interessata in una serie di piccoli trapezi aventi altezza SU e basi corrispondenti a $f(x)$ e $f(x+SU)$.

La loro area è

$$\frac{f(x) + f(x + SU)}{2} \times SU$$

cui si volesse procedere con un incremento più piccolo (per esempio 0.1) bisogna inserirlo al posto di 0.25 nelle righe 490 - 620 e 770.

Se si desidera ottenere tabulati piuttosto lunghi (superiori ad una pagina) è opportuno attivare la paginazione automatica della stampante aggiungendo nella riga 2030 un

```
PRINT#4, CHR$(147)
```

prima del RETURN.

Il comando di chiusura del canale di stampa è inutile perché se in 870 viene deciso di procedere ad un nuovo calcolo il comando RUN chiude automaticamente tutti i canali compreso quello di stampa.

Questo programma può essere rapidamente modificato per il Plus 4 e per il C16 sostituendo le righe 200 e 260 con le seguenti:

```
200 FOR K=0 TO 2:POKE 1319+K,13:NEXT:POKE 239,3
260 POKE 239,0
```

Le quattro figure seguenti rappresentano invece una uscita su stampante-




GRAFICO DELLA FUNZIONE :

$$Y = \sin(X)$$

ESTREMI DI VISUALIZZAZIONE :

```
X (SINISTRA)=-5
X (DESTRA)= 15
Y (INFERIORE)=-2
Y (SUPERIORE)= 2
```

ESTREMI DI INTEGRAZIONE :

```
X1 = 0
X2 = 3.14
```

Fig. 4.3




GRAFICO DELLA FUNZIONE :

$$Y = \text{SIN}(X)/X$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-20
 X (DESTRA)= 20
 Y (INFERIORE)=-.5
 Y (SUPERIORE)= 1.2

ESTREMI DI INTEGRAZIONE :

X1 = 3.14
 X2 = 6.28

Fig. 4.4

te del programma relativo al calcolo degli integrali, con il C64. (La modifica per Plus 4 e C16 verrà trattata nel capitolo successivo.) Si noti che nel grafico della funzione

$$y = \sin(x)/x$$

la discontinuità per $x=0$ può essere superata segnalando al calcolatore un inesistente asintoto verticale che evita l'errore per divisione per zero. La funzione successiva presenta l'elaborazione del calcolo seguente

$$\int_{\pi/6}^{\pi/3} \sin^2 x \cos x \, dx = \frac{3\sqrt{3} - 1}{24}$$

Gli estremi di integrazione sono stati forniti nella forma approssimata

$$X1 = .52$$

$$X2 = 1.04$$

Il risultato finale, anch'esso approssimato, è

AREA = .17




GRAFICO DELLA FUNZIONE :

$$Y = (\sin(X))^2 * \cos(X)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA)=-5
 X (DESTRA)= 10
 Y (INFERIORE)=-.8
 Y (SUPERIORE)= .8

ESTREMI DI INTEGRAZIONE :

X1 = .52
 X2 = 1.04

Fig. 4.5




GRAFICO DELLA FUNZIONE :

$$Y = (2 * X - 1) / (2 * X * 13)$$

ESTREMI DI VISUALIZZAZIONE :

X (SINISTRA) = -2
 X (DESTRA) = 4
 Y (INFERIORE) = -2
 Y (SUPERIORE) = 2

ESTREMI DI INTEGRAZIONE :

X1 = .5
 X2 = 4

Fig. 4.6

Grafici con il Plus 4 e C16

GRAFICO DI UNA FUNZIONE E CALCOLO DI UN'AREA

Il programma, con alcune modifiche, è identico a quello relativo al C64. Sono state modificate le POKE nelle righe 80 e 510 corrispondenti al buffer di tastiera e al relativo puntatore, in quanto situate in locazioni di memoria differenti.

Il programma risulta molto più breve (anche perché manca la routine di *hard-copy*) e gira anche sul C16 senza che sia necessario cancellare le REM per guadagnare memoria.

Una differenza molto importante rispetto al listato del C64, dovuta al Basic più evoluto, è la capacità di controllare gli errori (in particolare la divisione per zero e l'*overflow*) senza interrompere lo scorrimento del programma.

Nel caso si verifichi uno di questi errori il programma traccia automaticamente l'asintoto verticale che quindi non è più necessario dichiarare all'inizio.

```
10 REM *****
20 REM *      COPYRIGHT IN BASIC      *
21 REM *      VIA SEZZE 22 - LATINA    *
22 REM *      TEL. 0773 /487631      *
23 REM *-----*
24 REM *      STUDIO FUNZIONI E CALCOLO *
25 REM *      DI INTEGRALI            *
26 REM *      PER C-16 E PLUS-4      *
30 REM *****
40 COLOR 0,15,2:COLOR 4,7,5:COLOR 1,0,6
50 PRINT"UBATTI LA FUNZIONE :":INPUT"Y=";Y*
60 PRINT"UBATTI 500 DEF FNY(X)="Y*
```

```

70 PRINT"GOTO 500"
80 FOR K=0 TO 2:POKE 1319+K,13:NEXT:POKE 239,3
90 END
500 DEF FNY(X)=SIN(X)
510 POKE 239,0:PRINT"OK"
520 PRINT"SCGLI GLI INTERVALLI DI VISUALIZZAZIONE"
530 INPUT"X (SINISTRA)=":X1
540 INPUT"X (DESTRA)=":X2
545 IF X1>X2 THEN 530
550 INPUT"Y (BASSA)=":Y1
560 INPUT"Y (ALTA)=":Y2
570 FL=0:PRINT"VUOI CALCOLARE UN'AREA ?"
580 GETKEY A$
590 IF A$(">")="S" THEN 1000
600 FL=1:INPUT"ASCISSA INIZIALE=":A1
610 INPUT"ASCISSA FINALE=":AF
1000 REM -----
1010 REM          VISUALIZZAZIONE
1020 REM -----
1025 PRINT"U"
1030 DX=(X2-X1)/320:DY=(Y2-Y1)/200:GRAPHIC 1,1
1040 REM ---- ASSI CARTESIANI ----
1050 IF SGN(Y1)=SGN(Y2) THEN 1070
1060 Y=INT(Y2/DY+.5):DRAW 1,0,Y TO 319,Y
1065 CHAR 1,38,1+Y/8,"X"
1070 IF SGN(X1)=SGN(X2) THEN 1090
1080 X=-INT(X1/DX+.5):DRAW 1,X,0 TO X,199
1085 CHAR 1,1+X/8,1,"Y"
1090 FOR K=0 TO 319
1100 X=X1+K*DX:TRAP 2700
1110 Y=FNY(X):GOSUB 4000
1120 NEXT
1130 IF FL=0 THEN 2000
1140 REM -----
1150 REM          I N T E G R A L E
1160 REM -----
1170 SU=(AF-A1)/200
1180 FOR K=A1 TO AF-SU STEP SU
1190 IN=IN+(FNY(K)+FNY(K+SU))*SU/2
1200 NEXT K
1210 IN=INT(IN*100+.5)/100
1220 R$=STR$(IN):R$="AREA="+R$
1230 CHAR 1,1,1,R$
1240 FOR X0=A1 TO AF STEP DX
1250 X=(X0-X1)/DX
1260 AY=INT(Y2/DY+.5)
1270 YY=200-(FNY(X0)-Y1)/DY
1280 DRAW 1,X,AY TO X,YY
1290 NEXT X0
2000 GETKEY A$:GRAPHIC 0
2010 PRINT"ANCORA CON LA STESSA FUNZIONE ?":GETKEY A$
2020 IF A$="S" THEN RUN 500
2030 PRINT"CON UN' ALTRA FUNZIONE ?":GETKEY A$
2040 IF A$="S" THEN RUN
2050 PRINT"OK - FINE LAVORO."
2060 END
2500 REM -----
2510 REM TRACCIAM.ASINTOTO VERTICALE
2520 REM -----
2700 FOR J=0 TO 199 STEP 6:DRAW 1,K,J:NEXT
2710 TRAP:RESUME NEXT
4000 REM -----
4010 REM          PLOTTAMENTO
4020 REM -----
4050 IF Y<Y1 THEN 4100

```

```

4060 IF Y>=Y2 THEN 4100
4070 Y=200-(Y-Y1)/DY
4080 DRAW 1,K,Y
4100 RETURN

```

Scorriamo rapidamente il listato. Nella riga 40 vengono fissati i colori (e la luminanza) dello schermo, della cornice e dei caratteri.

In 500 il nuovo comando GET KEY A\$ corrisponde alla linea

```
580 GET A$:IF A$=""THEN 580
```

che genera un ciclo di attesa.

In 1030 viene attivata l'alta risoluzione con

```
GRAPHIC 1,1
```

mentre in 1060 e 1080 il comando

```
DRAW 1,a,b TO c,d
```

traccia una linea fra i punti di coordinate (a,b) e (c,d) che nel nostro caso sono gli assi cartesiani.

Nelle righe 1065 e 1085 vengono stampati i caratteri X e Y in corrispondenza degli assi.

Il comando che permette al calcolatore di tracciare gli asintoti verticali in corrispondenza delle divisioni per zero e degli *overflow* è il TRAP (riga 1100) che invia alla subroutine 2700.

Per il resto il programma è identico a quello già visto per il C64.

L'uscita su stampante MPS-802 può essere realizzata con una routine presa da una analoga di Danilo Toma pubblicata su *Computer Club*.

Per il C16 e il Plus 4 occorre modificare le righe 2000-2010 nel modo seguente

```

2000 GETKEY A$:IF A$(">"|"<") THEN 2010
2005 DRAW 1,0,0 TO 319,0:DRAW 1,0,199 TO 319,199
2007 DRAW 1,0,0 TO 0,199:DRAW 1,319,0 TO 319,199
2008 GOSUB 5000
2010 GRAPHIC 0:PRINT"ANCORA CON LA STESSA FUNZIONE ?":GETKEY A$

```

Inoltre si deve aggiungere la subroutine

```

5000 REM -----
5010 REM ROUTINE DI STAMPA PER MPS-802
5020 REM -----
5030 DATA 169,0,133,253,169,32,133,254,96,160,7,32,108,3,177
5040 DATA 253,153,124,3,136,16,248,32,116,3,160,7,162,7,94
5050 DATA 124,3,106,202,16,249,153,132,3,136,16,241,24,165
5060 DATA 253,105,0,133,253,165,254,105,0,133,254,96,120

```

```

5070 DATA 169,252,37,1,133,1,96,169,3,5,1,133,1,68,96
5080 RESTORE:TD=0:FOR K=820 TO 891
5090 READ N:TD=TD+N:POKE K,N
5100 NEXT
5110 IF TD<>7874 THEN END:REM ERRORE NEI DATA
5120 C$="" " :REM 20 SPAZI
5130 OPEN 4,4:OPEN 5,4,5:OPEN 6,4,6
5140 SYS 820
5150 FOR K=0 TO 24:PRINT#6,CHR$(K)
5160 B$=C$:FOR J=0 TO 39
5170 SYS 829
5180 A$="" :FOR T=900 TO 907:A$=A$+CHR$(PEEK(T)):NEXT T
5190 PRINT#5,A$
5200 PRINT#4,B$ :B$=B$+CHR$(32)
5210 PRINT#4,CHR$(254):NEXT J
5220 PRINT#6,CHR$(20):PRINT#4:NEXT K
5230 CLOSE 4:CLOSE 5:CLOSE 6
5240 OPEN 4,4,10:PRINT#4:CLOSE 4
5250 RETURN

```

La stessa routine può essere usata per la stessa stampante e il C64, con l'unica modifica di un DATA: il sesto DATA della riga 5030 deve essere trasformato da 32 in 224.

FUNZIONI POLIVOCHE O SOVRAPPOSTE CON IL C16 E IL PLUS 4

Anche questo programma è la trasposizione esatta del programma già visto per il C64. Le uniche modifiche riguardano i nuovi comandi per l'alta risoluzione e il riconoscimento automatico degli asintoti verticali.

```

10 REM *****
20 REM *   COPYRIGHT IN BASIC   *
21 REM *   VIA SEZZE 22 - LATINA *
22 REM *   TEL. 0773 /487631   *
23 REM *-----*
24 REM * STAMPA FUNZIONI POLIVOCHE *
25 REM *   PER C-16 E PLUS-4   *
30 REM *****
40 COLOR 0,15,2:COLOR 4,7,5:COLOR 1,0,6
50 PRINT"███ 1 - VISUALIZZAZIONE FUNZIONI"
60 PRINTTAB(5)*2 - CAMBIO DELLE FUNZIONI"
70 PRINTTAB(5)*3 - SCELTA DEGLI INTERVALLI"
75 PRINTTAB(5)*4 - USCITA PER FINE LAVORO"
80 GETKEY A$
85 IF A$="4" THEN PRINT"███ - FINE LAVORO." :END
90 IF ASC(A$)<49 OR ASC(A$)>51 THEN 80
100 A=ASC(A$)-48:ON A GOTO 1000,2000,3000
1000 REM -----
1010 REM   VISUALIZZAZIONE
1020 REM -----
1030 DX=(X2-X1)/320:DY=(Y2-Y1)/200:GRAPHIC 1,1
1040 REM ---- ASSI CARTESIANI ----
1050 IF SGN(Y1)=SGN(Y2) THEN 1070
1060 Y=INT(Y2/DY+.5):DRAW 1,0,Y TO 319,Y
1065 CHAR 1,38,1+Y/8,"X"
1070 IF SGN(X1)=SGN(X2) THEN 1090

```

```

1000 X=-INT(X1/DX+.5):DRAW 1,X,0 TO X,199
1005 CHAR 1,1+X/8,1,"Y"
1090 GOSUB 2490
1100 GETKEY A$:GRAPHIC 0
1110 GOTO 40
2000 REM -----
2010 REM CAMBIO DELLE FUNZIONI
2020 REM -----
2030 PRINT"-----"
2040 LIST 2500-2650:END
2490 FOR K=0 TO 319
2495 X=X1+K*DX:TRAP 2700
2500 Y=(X-2)/(X*X+X-2):GOSUB 4000
2510 :
2520 Y=(X*X+X+9)/(X*X-2*X-3):GOSUB 4000
2530 :
2540 :
2550 :
2560 :
2570 :
2580 :
2600 :
2610 :
2620 :
2630 :
2640 :
2650 REM"-----"RUN 40
2660 NEXT K
2670 RETURN
2700 FOR J=0 TO 199 STEP 6:DRAW 1,K,J:NEXT
2710 TRAP:RESUME NEXT
3000 REM -----
3010 REM SCELTA DEGLI INTERVALLI
3020 REM -----
3030 PRINT"SCGLI GLI INTERVALLI DI VISUALIZZAZIONE"
3040 INPUT"X (SINISTRA)":X1
3050 INPUT"X (DESTRA)":X2
3060 IF X1>X2 THEN 3030
3070 INPUT"Y (BASSA)":Y1
3080 INPUT"Y (ALTA)":Y2
3090 IF Y1>Y2 THEN 3070
3300 GOTO 40
4000 REM -----
4010 REM PLOTTAMENTO
4020 REM -----
4050 IF Y<=Y1 THEN 4100
4060 IF Y>=Y2 THEN 4100
4070 Y=200-(Y-Y1)/DY
4080 DRAW 1,K,Y
4100 RETURN

```

FUNZIONI IN COORDINATE POLARI

Un punto nel piano può essere individuato con sistemi di riferimento diversi da quello cartesiano. Per esempio, basta fissare un punto O e una semiretta r uscente da questo punto. La posizione di un altro punto P




Fig. 5.1

qualsiasi del piano può essere allora determinata misurando la distanza P da O e l'angolo che il segmento PO forma con la semiretta r . Il segmento r si chiama modulo, mentre l'angolo t si chiama *anomalia*. Immaginando una sovrapposizione fra tale sistema di riferimento e una coppia di assi cartesiani con l'origine coincidente con O e l'asse r sovrapposto a x , si possono elaborare formule di trasformazione da un sistema all'altro.




Fig. 5.2

$$y = r \sin(t)$$

$$x = r \cos(t)$$

Quadrando e sommando si ha:

$$y^2 = r^2 (\sin(t))^2$$

$$x^2 = r^2 (\cos(t))^2$$

$$x^2 + y^2 = r^2 [(\sin(t))^2 + (\cos(t))^2]$$

$$x^2 + y^2 = r^2$$

$$r = \sqrt{x^2 + y^2}$$

mentre invece dividendo membro a membro

$$y/x = \text{tang}(t)$$

$$t = \text{arctang}(y/x)$$

Talvolta conviene scrivere l'equazione di una curva utilizzando il sistema di riferimento polare, perché l'equazione stessa assume una forma più semplice. Alcuni esempi:

$r = \text{costante}$	equazione di una circonferenza con centro nel polo
$t = \text{costante}$	equazione di una retta con centro nel polo
$r = -8 \sin(t)$	con le formule di trasformazione si ottiene $x^2 + y^2 + 8y = 0$ cioè una circonferenza con centro nel punto $(0; -4)$ e raggio 4
$r = 4 \sin(t) / \cos(t)$	equazione di una parabola con centro nell'origine e asse coincidente con l'asse y
$r = 30 (1 - \cos(t))$	la curva rappresentata è una cardioide
$r = 1/(\sin(t) - 2\cos(t))$	equazione di una retta avente in un riferimento cartesiano la forma $y = 2x + 1$
$r = 6 t/\pi$	spirale di Archimede
$r = 30 \sin(3t)$	rosa a tre foglie con centro di simmetria nell'origine

$r = 20 / \sqrt{(\cos(t))^2 - (\sin(t))^2}$ equazione di un'ellisse avente nel piano
 cartesiano la forma
 $x^2 / 400 + y^2 / 100 = 1$

Il programma per visualizzare tali funzioni è il seguente:

```

10 REM *****
20 REM *      COPYRIGHT IN BASIC      *
30 REM *      VIA SEZZE 22 - LATINA   *
40 REM *      TEL. 0773 /487631      *
50 REM *-----*
60 REM * FUNZIONI IN COORDINATE POLARI *
70 REM *****
80 COLOR 0,15,2:COLOR 4,7,5:COLOR 1,8,6
90 PRINT"␣"
100 PRINT"LA FUNZIONE DEVE AVERE L'ANGOLO T COME"
110 PRINT"VARIABILE INDIPENDENTE"
120 INPUT"ANGOLO R =";Y$
130 PRINT"ANGOLO 220 R="Y$
140 PRINT"RUN 170"
150 FOR K=0 TO 2:POKE 1319+K,13:NEXT:POKE 239,3
160 END
170 POKE 239,0:PRINT"ANGOLO FATTORE DI SCALA ?"
175 PRINT"10/1 0 1/1 ? (BATTI 1 O 2)"
177 GETKEY A$:IF A$<"1" OR A$>"2" THEN 177
178 A=10:IF A$="2" THEN A=1
180 GRAPHIC 1,1
185 DRAW 1,0,100 TO 319,100
167 DRAW 1,160,0 TO 160,199
190 DO
200 LOCATE 160,100:TRAP 250
210 REM ----- FUNZIONE -----
220 R=30/SQR(COS(T)^2+4*SIN(T)^2)
230 REM -----
235 IF R<0 THEN 250
240 LOCATE R*A/450-G:DRAW 1,RD0T(0),RD0T(1)
250 G=G+1:T=G*PI/180
260 R=INT(R*1000+.5)/1000
270 CHAR1,2,2,"R="+STR$(R)+" "
280 CHAR1,2,3,"T="+STR$(G)
290 LOOP UNTIL G=360
300 GETKEY A$
310 GRAPHIC 0
320 PRINT"ANCORA CON LA STESSA FUNZIONE ?"
330 GETKEY A$
340 IF A$="S" THEN RUN 170
350 PRINT"CON UNA FUNZIONE DIFFERENTE ?"
360 GETKEY A$
370 IF A$="S" THEN RUN
380 PRINT"OK - FINE LAVORO."

```

Il Plus 4 e il C16 hanno la possibilità di controllare e modificare la posizione del cursore anche in alta risoluzione. In tal caso prende il

nome di *pixel cursore* e l'unica differenza rispetto al cursore in bassa risoluzione è rappresentata dal fatto che non lampeggia. È però di notevole importanza avere la possibilità di posizionare il pixel cursore nel punto voluto dello schema, prima di iniziare un tracciamento. Il comando relativo è

LOCATE X,Y

dove X ed Y sono le coordinate del punto in cui vogliamo indirizzare il pixel cursore.

Vi è la possibilità di usare tale istruzione nella forma

LOCATE a;b

dove a rappresenta la lunghezza e b l'angolo (in gradi) dello spostamento desiderato.

I cicli FOR...NEXT possono essere sostituiti con la coppia di istruzioni

DO LOOP UNTIL...

che significa testualmente

ESEGUI RIPETI FINO A QUANDO...

Nel listato, l'introduzione della funzione avviene con il solito inserimento automatico nella riga 220.

Nell'intervallo tra 190 e 290 viene attivato un ciclo DO...LOOP UNTIL... in cui la variabile T (espressa in radianti) e la variabile G (espressa in gradi) compiono una rotazione completa.

Nella riga 200, a ogni ciclo il pixel cursore viene posizionato al centro del video. Il TRAP impedisce l'interruzione dell'elaborazione nel caso in cui vi sia un errore nella riga successiva.

In 240 il pixel cursore viene indirizzato nel punto calcolato nella riga 220.

I comandi RDOT(0) e RDOT(1) servono a conoscere la posizione del pixel cursore (in coordinate cartesiane relative allo schermo), mentre il comando

DRAW 1, RDOT(0), RDOT(1)

serve ad accendere il pixel interessato.

In 250 si attuano l'incremento della variabile G e la sua trasformazione in radianti.

Si noti come sia stato necessario calcolare il complemento a 450 di ogni angolo prima del tracciamento, per compensare il differente metodo di determinazione degli angoli fra il sistema americano ed il nostro: in quello usato dal calcolatore gli angoli vengono contati a cominciare dalle ore 12 e procedono in senso orario.

Infine in 270-280 si ha la visualizzazione dei valori di R e T calcolati via via durante l'elaborazione.

Si noti che in 235 viene imposto un salto che evita l'accensione del pixel nel caso in cui il modulo sia negativo.

NUMERI COMPLESSI E LORO RAPPRESENTAZIONE NEL PIANO DI GAUSS

Dato un generico numero complesso

$$Z = a + i b$$

possiamo rappresentarlo in un sistema di assi cartesiani, considerando il coefficiente reale a come ascissa di un punto e il coefficiente immaginario b come ordinata.

In tal caso il piano prende il nome di *piano di Gauss*, l'asse delle ascisse viene denominato *asse reale* e l'asse delle ordinate *asse immaginario*. Ogni numero complesso equivale quindi a un punto sul piano di Gauss e talvolta conviene far corrispondere a tale numero complesso il vettore che parte dall'origine e termina nel punto di coordinate (a, b) .

Considerando il numero complesso come un vettore possiamo applicare a esso le considerazioni già svolte nel paragrafo precedente e determinare per ciascuno di essi i corrispondenti valori del modulo e dell'anomalia.

Sui numeri complessi (e sui vettori) è possibile eseguire le quattro operazioni aritmetiche e, calcolato il risultato, visualizzarlo sotto forma di vettore.

Con un programma di questo tipo è possibile mostrare l'applicazione della *regola del parallelogramma* sia per la somma sia per la differenza e, nel caso di prodotto o quoziente per un numero immaginario, la rotazione del vettore risultante.

```

10 REM *****
20 REM *      COPYRIGHT IN BASIC      *
30 REM *      VIA SEZZE 22 - LATINA   *
40 REM *      TEL. 0773 /487631      *
50 REM *-----*
60 REM * PIANO DI GAUSS (C-16 PLUS-4) *
70 REM *****
80 COLOR 0,15,2:COLOR 4,7,5:COLOR 1,8,6
90 PRINT"␣":GRAPHIC 2,1
95 REM -----
96 REM      TRACCIAMENTO ASSI
97 REM -----
100 DRAW 1,0,80 TO 319,80
103 FOR K=0 TO 319 STEP 10:FOR J=78 TO 82
104 DRAW 1,K,J:NEXT J,K
110 CHAR 1,33,11,"R"
120 DRAW 1,160,0 TO 160,160
123 FOR K=0 TO 159 STEP 10:FOR J=158 TO 162
124 DRAW 1,J,K:NEXT J,K
130 CHAR 1,18,1,"I"
140 DRAW 1,0,159 TO 319,159
150 REM -----
160 REM      INGRESSO DATI
170 REM -----
175 PRINT"SCALA = 1/10 0 1/1 ? BATTI 0 1 2 ";
177 GETKEY A$:IF A$(">")="1" AND A$(">")="2" THEN 177
178 PRINTA$:FM=1:IF A$="2" THEN FM=10
190 INPUT"PARTE REALE PRIMO VETTORE":R1
200 INPUT"PARTE IMMAGINARIA":I1
210 R=R1:I=I1:GOSUB 2000
215 M1=M:A1=A:GOSUB 3000
220 INPUT"PARTE REALE SECONDO VETTORE":R2
230 INPUT"PARTE IMMAGINARIA":I2
240 R=R2:I=I2:GOSUB 2000
245 M2=M:A2=A:GOSUB 3000
250 REM -----
260 REM      SCELTA DEL CALCOLO
270 REM -----
280 PRINT" SCEGLI IL TIPO DI OPERAZIONE : "
290 PRINTTAB(12)" + - * / "
300 GETKEY A$
310 IF A$="+" OR A$="-" OR A$="*" OR A$="/" THEN 330
320 GOTO 300
330 IF A$="+" THEN GOSUB 4000:GOTO 500
340 IF A$="-" THEN GOSUB 5000:GOTO 500
350 IF A$="*" THEN GOSUB 6000:GOTO 500
360 GOSUB 7000
370 REM -----
380 REM      R I S U L T A N T E
390 REM -----
500 GOSUB 2000:GOSUB 3000
510 PRINT" IL VETTORE RISULTANTE HA : "
515 R=INT(R*1000+.5)/1000:I=INT(I*1000+.5)/1000
520 PRINT" P. REALE="R;TAB(18)"P. IMMAGINARIA="I
530 PRINT" MODULO="M;TAB(18)"ANOMALIA="450-A
540 GETKEY A$
550 PRINT" ANCORA ? "
560 GETKEY A$
570 IF A$="S" THEN RUN
580 PRINT"␣":GRAPHIC 0
590 PRINT"OK - FINE LAVORO." :END
2000 REM -----
2010 REM      CALCOLO MODULO E ANOMALIA
2020 REM -----
2030 M=SQR(R^2+I^2):M=INT(M*100+.5)/100

```

```

2035 IF R=0 AND I>0 THEN A=90:GOTO 2060
2037 IF R=0 AND I<0 THEN A=270:GOTO 2060
2040 A=ATN(I/R)*180/PI:A=INT(A+.5)
2045 IF R<0 THEN A=A+180
2050 IF A>360 THEN A=A-360:GOTO 2050
2055 IF A<0 THEN A=A+360:GOTO 2055
2060 A=450-A:RETURN
3000 REM -----
3010 REM          TRACCIAMENTO VETTORE
3020 REM -----
3025 DRAW 1,161,79
3040 LOCATE M*FM/A
3050 X1=RDOT(0):Y1=RDOT(1)
3055 IF X1<0 OR X1>319 THEN 3070
3057 IF Y1<0 OR Y1>199 THEN 3070
3060 DRAW 1,161,79 TO X1,Y1
3070 RETURN
4000 REM -----
4010 REM          S O M M A
4020 REM -----
4030 R=R1+R2:I=I1+I2:RETURN
5000 REM -----
5010 REM          S O T T R A Z I O N E
5020 REM -----
5030 R=R1-R2:I=I1-I2:RETURN
6000 REM -----
6010 REM          P R O D D O T T O
6020 REM -----
6030 M=M1*M2:A=A1+A2
6035 R=R1*R2-I1*I2:I=R1*I2+R2*I1
6040 RETURN
7000 REM -----
7010 REM          Q U O Z I E N T E
7020 REM -----
7030 IF M2=0 THEN M=1000:GOTO 7045
7040 M=M1/M2:A=A1-A2
7045 DD=I2*I2+R2*R2
7047 R=(R1*R2+I1*I2)/DD:I=(I1*R2-R1*I2)/DD
7050 RETURN

```

Con il comando GRAPHIC 2,1 viene attivata l'alta risoluzione su un reticolato di 320x160 pixel, in quanto le ultime cinque righe in basso nello schermo rimangono in bassa risoluzione e possono essere sfruttate per la visualizzazione dei messaggi necessari durante lo scorrimento del programma.

Dopo il tracciamento degli assi nella routine 150-245 si ha l'introduzione dei coefficienti (reale e immaginario) dei due numeri complessi proposti.

La subroutine 2000 esegue la trasformazione calcolando il modulo e l'anomalia dei vettori corrispondenti, mentre la subroutine 3000 traccia il vettore stesso.

Nella routine 250-360 si ha la scelta dell'operazione e lo smistamento della subroutine interessata.

Infine in 370-590 viene visualizzato il vettore risultante.

Eliminando tutti i REM il programma gira anche sul C16. In tal caso occorre cambiare i seguenti numeri di linea

2030 va sostituito con **2000**
3025 va sostituito con **3000**
4030 va sostituito con **4000**
5030 va sostituito con **5000**
6030 va sostituito con **6000**
7030 va sostituito con **7000**



*Finito di stampare nel settembre 1985
presso la Lito Velox - Trento
Printed in Italy*

Il libro affronta i problemi legati alla realizzazione di grafici delle funzioni matematiche: spiega linea dopo linea i criteri che guidano la scrittura di programmi adeguati per visualizzare funzioni di ogni tipo.

Dalle scuole medie all'università, tutti gli studenti troveranno in questo libro come ottenere grafici in alta risoluzione (64.000 punti) e come stamparli su carta per mezzo di una comune stampante Commodore (MPS/801 o MSP/802).

I programmi girano sia sul Commodore 64 sia su C16 e PLUS4, sfruttando il Basic più evoluto di queste macchine.



franco muzzio & c. editore

ISBN 88-7021-332-3

L. 15.000 (14.706)